code:
@COM148
;穿环
CALL PRINT_TRAIN_NAME(SELECTCOM)

;-------------------------------------------------
;実行できるかの判定
;-------------------------------------------------
CALL SHOW_COM_ORDER_COM, 148

;実行できない
SIF TCVAR:干劲 == 0
	RETURN 0

;-------------------------------------------------
;実行決定
;-------------------------------------------------
CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

RETURN 1

@FLAG_COM148, ARG
;性器付近の穿環なら、阴部を露出させる
IF COND("陰蒂環フラグ") + COND("陰唇環フラグ") + COND("尿道穿環フラグ") + COND("恥丘穿環フラグ")
	CALL SET_BODYS, ""
	CALL SET_OUTER, ""
	CALL SET_SKIRT, ""
	SIF CHECK_PANTIES("開放型開洞") == 0
		CALL SET_PANTIES, ""
ENDIF

STAIN:PLAYER:手 = 0
IF COND("陰蒂環フラグ") || COND("陰唇環フラグ") || COND("尿道穿環フラグ") || COND("恥丘穿環フラグ")
	STAIN:Ｖ = SET_STAIN(0, "Ｖ")
ELSEIF COND("乳环フラグ")
	STAIN:Ｂ = 0
ENDIF

@SOURCE_COM148, ARG
;外した場合
SIF COND("穿環ＯＮフラグ") == 0
	RETURN 1

CALL SOURCE_SMPLAY, "穿环", ARG

CALL VARSET_COMF

CALL SOURCE_DOWNBASE_HP, 148

;基本は肚脐环の時
;初めてつける時は疼痛のせいで、体力消費が凄いことになります

SOURCE:0:疼痛 = 3000
SOURCE:0:屈従 = 1000
SOURCE:0:逸脱 = 3000

IF COND("陰蒂環フラグ")
	SOURCE:0:疼痛 += 17000
	SOURCE:0:屈従 += 9000
	SOURCE:0:逸脱 += 17000
ELSEIF COND("陰唇環フラグ") || COND("尿道穿環フラグ")
	SOURCE:0:疼痛 += 7000
	SOURCE:0:屈従 += 9000
	SOURCE:0:逸脱 += 12000
ELSEIF COND("乳环フラグ")
	SOURCE:0:疼痛 += 12000
	SOURCE:0:屈従 += 9000
	SOURCE:0:逸脱 += 12000
ELSEIF COND("恥丘穿環フラグ")
	SOURCE:0:疼痛 += 2000
	SOURCE:0:屈従 += 5000
	SOURCE:0:逸脱 += 7000
ELSEIF COND("肚脐环フラグ")

ENDIF

;ABL:受虐属性をみる
SELECTCASE ABL:受虐属性
CASE 0
CASE 1
	TIMES SOURCE:0:屈従, 1.20
	TIMES SOURCE:0:逸脱, 0.90
CASE 2
	TIMES SOURCE:0:屈従, 1.40
	TIMES SOURCE:0:逸脱, 0.80
CASE 3
	TIMES SOURCE:0:屈従, 1.60
	TIMES SOURCE:0:逸脱, 0.50
CASE 4
	TIMES SOURCE:0:屈従, 1.80
	TIMES SOURCE:0:逸脱, 0.30
CASEELSE
	TIMES SOURCE:0:屈従, 2.00
	TIMES SOURCE:0:逸脱, 0.10
ENDSELECT

;麻酔を使ったら疼痛激減
SIF COND("鎮痛麻酔フラグ")
	TIMES SOURCE:0:疼痛, 0.01

;视频拍摄中
IF TEQUIP:攝像機
	TIMES SOURCE:0:屈従, 1.50
	TIMES SOURCE:0:逸脱, 1.50
ENDIF

CALL CALC, "調教ソース", ARG

@TRAIN_MESSAGE_COM148
CALL SET_PIERCE, TARGET
;-------------------------------------------------
;ソースの計算(確定値)
;-------------------------------------------------
CALL CALC, "調教ソースの計算"

;地の文スキップは不可能としておく
@SET_PIERCE, ARG
;まずは穿環関係のフラグに重置をかける
TCVAR:ARG:イベント = CLEARBITS(TCVAR:ARG:イベント, 11, 20)

CALL SET_PIERCE_MAIN, ARG

@SET_PIERCE_MAIN, ARG
#DIM MEMO_LINECOUNT
#DIM LCOUNT
;穿環の形状
#DIMS PIERCES
;穿環の場所
#DIMS PART_PIERCES
#DIM CHOICE

PART_PIERCES =

REDRAW 0
MEMO_LINECOUNT = LINECOUNT

IF ARG != MASTER && ABL:ARG:受虐属性 < 3 && FIRSTCOM(SELECTCOM, ARG) == 0
	PRINTFORML 這種激烈的PLAY對她來說還太早了點吧？……也好。這次、又要怎麼做呢……？
ELSE
	PRINTFORML 所以、這次要怎麼做呢……？
ENDIF

IF COND("陰蒂環可能", ARG) == 0
	CALL PRINT_STRL, @"灰色_ [×] \@ PENIS(ARG) ? 阴茎穿環　　 # 阴蒂穿環 \@　（要:[淫核]or[淫乱]or[倒錯的]or受虐属性3LV）"
ELSE
	PRINTFORML  [ 0] 將\@ PENIS(ARG) ? 阴茎 # 阴蒂 \@\@ TALENT:ARG:陰蒂環 ? 上的穿環取下 # 用金屬穿環装飾起來 \@
ENDIF

IF TALENT:ARG:男性
	CALL PRINT_STRL, @"灰色_ [×] 小陰唇穿環　　　（※男性是不行的※）"
ELSEIF COND("陰唇環可能", ARG) == 0
	CALL PRINT_STRL, @"灰色_ [×] 小陰唇穿環　　　（要:[淫壷]or[淫乱]or[倒錯的]or受虐属性2LV）"
ELSE
	PRINTFORML  [ 1] 將小陰唇\@ TALENT:ARG:陰唇環 ? 上的穿環取下 # 用金屬穿環装飾起來 \@
ENDIF

IF COND("恥丘穿環可能", ARG) == 0
	CALL PRINT_STRL, @"灰色_ [×] 將恥丘用金屬穿環装飾起來　　　　（要:[淫核]or[淫乱]or[倒錯的]or受虐属性1LV）"
ELSE
	SELECTCASE CONDS("陰毛")
	CASE "白虎", "無毛", "脱毛処理", "光滑无毛"
		PRINTFORML  [ 2] 將恥丘\@ TALENT:ARG:恥丘穿環 ? 上的穿環取下 # 用金屬穿環装飾起來 \@　　　　（着ける前に剃毛を行います）
	CASEELSE
		PRINTFORML  [ 2] 將恥丘\@ TALENT:ARG:恥丘穿環 ? 上的穿環取下 # 用金屬穿環装飾起來 \@
	ENDSELECT
ENDIF

IF COND("真性陥没乳頭", ARG)
	CALL PRINT_STRL, @"灰色_ [×] 乳頭穿環　　　　（※真性陥没乳頭是不行的※）"
ELSEIF COND("乳环可能", ARG) == 0
	CALL PRINT_STRL, @"灰色_ [×] 乳頭穿環　　　　（要:[淫乳]or[淫乱]or[倒錯的]or受虐属性1LV）"
ELSE
	PRINTFORML  [ 3] 將乳頭\@ TALENT:ARG:乳环 ? 上的穿環取下 # 用金屬穿環装飾起來 \@
ENDIF

IF COND("尿道穿環可能", ARG) == 0
	CALL PRINT_STRL, @"灰色_ [×] 尿道口穿環　　　（要:尿道拡張0LVに加えて[尿道狂]or[淫乱]or[倒錯的]or受虐属性4LV）"
ELSE
	PRINTFORML  [ 4] 將尿道口\@ TALENT:ARG:尿道穿環 ? 上的穿環取下 # 用金屬穿環装飾起來 \@
ENDIF

PRINTFORML  [ 5] 將肚臍\@ TALENT:ARG:肚脐环 ? 上的穿環取下 # 用金屬穿環装飾起來 \@
PRINTFORML [100] 不、再考慮一下吧。

CALL INPUT_SELECT, 6, 100

SELECTCASE RESULT
CASE 100
	TFLAG:指令中止 = 1
	REDRAW 1
	RETURN 0
CASE 0
	IF COND("陰蒂環可能", ARG)
		CALL SETFLAG, "陰蒂環フラグ", ARG
		SIF TALENT:ARG:陰蒂環 == 0
			CALL SETFLAG, "穿環ＯＮフラグ", ARG
		CALL DEL_CEVENT_GROUP, "陰蒂環", ARG
		IF PENIS(ARG)
			PART_PIERCES = 阴茎
		ELSE
			PART_PIERCES = 阴蒂
		ENDIF
	ENDIF
CASE 1
	IF TALENT:ARG:男性
		PRINTFORMW ※男性怎么想都不可能吧※
	ELSEIF COND("陰唇環可能", ARG)
		CALL SETFLAG, "陰唇環フラグ", ARG
		SIF TALENT:ARG:陰唇環 == 0
			CALL SETFLAG, "穿環ＯＮフラグ", ARG
		CALL DEL_CEVENT_GROUP, "陰唇環", ARG
		PART_PIERCES = 小陰唇
	ENDIF
CASE 2
	IF COND("恥丘穿環可能", ARG)
		CALL SETFLAG, "恥丘穿環フラグ", ARG
		SIF TALENT:ARG:恥丘穿環 == 0
			CALL SETFLAG, "穿環ＯＮフラグ", ARG
		CALL DEL_CEVENT_GROUP, "恥丘穿環", ARG
		PART_PIERCES = 恥丘
	ENDIF
CASE 3
	IF COND("真性陥没乳頭", ARG)
		PRINTFORMW ※真性陥没乳頭无法装着※
	ELSEIF COND("乳环可能", ARG)
		CALL SETFLAG, "乳环フラグ", ARG
		SIF TALENT:ARG:乳环 == 0
			CALL SETFLAG, "穿環ＯＮフラグ", ARG
		CALL DEL_CEVENT_GROUP, "乳环", ARG
		PART_PIERCES = 乳頭
	ENDIF
CASE 4
	IF ABL:ARG:尿道拡張 < 0
		PRINTFORMW ※需要尿道拡張LV0以上才可以装着※
	ELSEIF COND("尿道穿環可能", ARG)
		CALL SETFLAG, "尿道穿環フラグ", ARG
		SIF TALENT:ARG:尿道穿環 == 0
			CALL SETFLAG, "穿環ＯＮフラグ", ARG
		CALL DEL_CEVENT_GROUP, "尿道穿環", ARG
		PART_PIERCES = 尿道
	ENDIF
CASE 5
	CALL SETFLAG, "肚脐环フラグ", ARG
	SIF TALENT:ARG:肚脐环 == 0
		CALL SETFLAG, "穿環ＯＮフラグ", ARG
	CALL DEL_CEVENT_GROUP, "肚脐环", ARG
	PART_PIERCES = 肚脐
ENDSELECT
IF PART_PIERCES == ""
	CLEARLINE LINECOUNT - MEMO_LINECOUNT
	RESTART
ENDIF

$INPUT_LOOP

;穿環の詳細を文字列で記録
PIERCES =

;穿環をつけた
IF COND("穿環ＯＮフラグ", ARG)
	;性器付近の穿環なら、阴部を露出させる
	IF COND("陰蒂環フラグ", ARG) + COND("陰唇環フラグ", ARG) + COND("尿道穿環フラグ", ARG) + COND("恥丘穿環フラグ", ARG) && !CHECK_CLO("Ｖ露出", ARG)
		CALL SET_PANTIES, "", ARG
		CALL SET_BODYS, "", ARG
		CALL SET_OUTER, "", ARG
	ENDIF

	SELECTCASE PART_PIERCES
	CASE "恥丘"
		SELECTCASE CONDS("陰毛")
		CASE "白虎", "無毛", "脱毛処理", "光滑无毛"
		CASEELSE
			BASE:陰毛 = 300
		ENDSELECT

		PRINTFORML 用什么颜色的穿環呢？
		PRINTFORML  [ 0] 銀色
		PRINTFORML  [ 1] 金色
		PRINTFORML  [ 2] 赤
		PRINTFORML  [ 3] 緑
		PRINTFORML  [ 4] 紫
		PRINTFORML  [ 5] 青
		PRINTFORML [100] 不拘泥於特定的形態（通常無記述）

		CALL INPUT_SELECT, 6, 100

		SELECTCASE RESULT
		CASE 100
		CASE 0
			PIERCES = 銀色
		CASE 1
			PIERCES = 金色
		CASE 2
			PIERCES = 赤
		CASE 3
			PIERCES = 緑
		CASE 4
			PIERCES = 紫
		CASE 5
			PIERCES = 青
		ENDSELECT

		CALL SET_CEVENT, @"恥丘穿環：%PIERCES%", ARG
		TALENT:ARG:恥丘穿環 = 1
		CALL PRINT_CUTIN, "すじ", ARG
		PRINTFORML ……装上的预想图大概是这个样子、这样可以吗？
		CALL PRINT_SELECT, "ＯＫ！/重新选择"
		CHOICE = RESULT

		CALL DEL_CEVENT, @"恥丘穿環：%PIERCES%", ARG
		TALENT:ARG:恥丘穿環 = 0
		SIF CHOICE
			GOTO INPUT_LOOP
	CASEELSE
		PRINTFORML 這次、要安装上那種穿環呢？
		PRINTFORML  [ 0] 圓環型
		PRINTFORML  [ 1] 啞鈴型
		PRINTFORML  [ 2] 正十字型
		PRINTFORML  [ 3] ♂型
		PRINTFORML  [ 4] ♀型
		PRINTFORML  [90] 直接輸入
		PRINTFORML [100] 不拘泥於特定的形態（通常無記述）

		CALL INPUT_SELECT, 5, 90, 100

		SELECTCASE RESULT
		CASE 100
		CASE 0
			PIERCES = 圓環型
		CASE 1
			PIERCES = 啞鈴型
		CASE 2
			PIERCES = 十字架型
		CASE 3
			PIERCES = ♂型
		CASE 4
			PIERCES = ♀型
		CASE 90
			PRINTL 請將全角８文字（半角則16文字）以下的字符串輸入。
			PRINTL (像是"耳環"之類被使用也不會產生不協感的記錄是推薦選項。)
			PRINTL 另外、如果是什麼也不輸入的場合就會被視作不算特別拘泥的場合（也即是無記述）。
			CALL INPUTS_SELECT, 16
			PIERCES = %RESULTS%
		ENDSELECT
	ENDSELECT

	IF ARG != MASTER
		PRINTL 使用麻醉劑嗎？
		PRINTL [0] 是
		PRINTL [1] 否
		CALL INPUT_SELECT, 2
		SIF RESULT == 0
			CALL SETFLAG, "鎮痛麻酔フラグ", ARG
	ENDIF

	IF ARG == MASTER
		PRINTFORM %CALLNAME:MASTER%自己的
	ELSE
		PRINTFORM %CALLNAME:PLAYER%将%CALLNAME:ARG%の
	ENDIF
	IF COND("陰蒂環フラグ", ARG)
		IF PENIS(ARG)
			PRINT 陰莖
		ELSE
			IF COND("陰核肥大", ARG)
				PRINT 腫脹肥大的
			ELSEIF TALENT:ARG:淫核
				PRINT 興奮勃起的
			ELSEIF TALENT:ARG:Ｃ性向
				PRINT 敏感而淫亂的
			ELSEIF TALENT:ARG:娇小
				PRINT 小小的
			ENDIF
			PRINT 阴蒂
		ENDIF
	ELSEIF COND("陰唇環フラグ", ARG)
		IF TALENT:ARG:淫壷
			PRINT 淫亂地顫抖著的
		ELSEIF TALENT:ARG:Ｖ性向
			PRINT 敏感而淫亂的
		ELSEIF AGE(ARG) == 1
			PRINT 幼嫩的
		ELSEIF TALENT:ARG:娇小
			PRINT 小小的
		ENDIF
		PRINT 小陰唇
	ELSEIF COND("恥丘穿環フラグ", ARG)
		PRINT 恥丘の下端に
	ELSEIF COND("乳环フラグ", ARG)
		IF COND("陥没乳頭", ARG)
			PRINTFORM 陷沒在乳肉中的\@ COND("乳頭肥大", ARG) ? 碩大的 # \@乳頭剝出
		ELSE
			IF COND("乳頭肥大", ARG)
				PRINT 腫脹肥大的
			ELSEIF TALENT:ARG:淫乳
				PRINT 興奮勃起的
			ELSEIF TALENT:ARG:Ｂ性向
				PRINT 敏感而淫亂的
			ELSEIF AGE(ARG) == 1
				PRINT 幼嫩的
			ELSEIF TALENT:ARG:娇小
				PRINT 小小的
			ENDIF
			PRINT 乳頭
		ENDIF
	ELSEIF COND("尿道穿環フラグ", ARG)
		PRINTFORM 広開的尿道口
	ELSEIF COND("肚脐环フラグ", ARG)
		PRINT 肚臍
	ENDIF
	SELECTCASE PIERCES
	CASE ""
	CASE "赤", "青"
		PRINTFORM %PIERCES%い
	CASEELSE
		PRINTFORM %PIERCES%の
	ENDSELECT
	PRINTFORMW 穿環装上了。

	IF COND("陰蒂環フラグ", ARG)
		IF PENIS(ARG)

		ELSE
			PRINTFORMW %CALLNAME:ARG%の\@ COND("陰核肥大", ARG) ? 大号的 # \@阴蒂は穿環を付けられたおかげで皮が捲られ、常時剥き出しの状態になってしまった…
		ENDIF
		PIERCES = 陰蒂環：%PIERCES%
	ELSEIF COND("陰唇環フラグ", ARG)
		SIF AGE(ARG) >= 3
			PRINTFORMW %CALLNAME:ARG%の小陰唇は付けられた穿環の重みに引っ張られ、一目で分かるほどに大陰唇からはみ出てしまっている…
		PIERCES = 陰唇環：%PIERCES%
	ELSEIF COND("恥丘穿環フラグ", ARG)
		PIERCES = 恥丘穿環：%PIERCES%
	ELSEIF COND("乳环フラグ", ARG)
		SIF COND("陥没乳頭", ARG)
			PRINTFORMW %CALLNAME:ARG%の乳頭は通された穿環のおかげで戻れなくなり、常時露出するように矯正されてしまった…
		PIERCES = 乳环：%PIERCES%
	ELSEIF COND("尿道穿環フラグ", ARG)
		PIERCES = 尿道穿環：%PIERCES%
	ELSEIF COND("肚脐环フラグ", ARG)
		PIERCES = 肚脐环：%PIERCES%
	ENDIF

	;穿環の形状フラグセット
	SIF PIERCES != ""
		CALL SET_CEVENT, PIERCES, ARG
;穿環を外した
ELSE
	IF ARG == MASTER
		PRINTFORMW %CALLNAME:MASTER%将自己%PART_PIERCES%上的穿環摘了下来
	ELSE
		PRINTFORMW %CALLNAME:PLAYER%将%CALLNAME:ARG%的%PART_PIERCES%上的穿環摘了下来
	ENDIF
ENDIF

;つけたり外したり
IF COND("陰蒂環フラグ", ARG)
	INVERTBIT TALENT:ARG:陰蒂環, 0
ELSEIF COND("陰唇環フラグ", ARG)
	INVERTBIT TALENT:ARG:陰唇環, 0
ELSEIF COND("恥丘穿環フラグ", ARG)
	INVERTBIT TALENT:ARG:恥丘穿環, 0
ELSEIF COND("乳环フラグ", ARG)
	INVERTBIT TALENT:ARG:乳环, 0
ELSEIF COND("尿道穿環フラグ", ARG)
	INVERTBIT TALENT:ARG:尿道穿環, 0
ELSEIF COND("肚脐环フラグ", ARG)
	INVERTBIT TALENT:ARG:肚脐环, 0
ENDIF

@COM_ORDER_COM148, ARG
VARSET LOCAL
LOCALS =

;すべての命令に共通の要素を考慮
;(信頼が高いと命令に従いやすいなど)
CALL COM_ORDER, ARG

LOCAL = RESULT

IF COND("疲労")
	LOCAL -= 40
	LOCALS = %LOCALS% - 疲労困憊(40)
ENDIF

SELECTCASE TALENT:有男朋友
CASE 2
	LOCAL -= 40
	LOCALS = %LOCALS% - %TALENT_NAME(GETNUM(TALENT, "有男朋友"), TALENT:有男朋友)%(40)
CASE 1
	LOCAL -= 30
	LOCALS = %LOCALS% - %TALENT_NAME(GETNUM(TALENT, "有男朋友"), TALENT:有男朋友)%(30)
ENDSELECT

;ABL:露出癖
IF ABL:露出癖
	LOCAL += MIN(ABL:露出癖, 10) * 2
	LOCALS = %LOCALS% + 露出癖LV{ABL:露出癖}({MIN(ABL:露出癖, 10) * 2})
ENDIF
;ABL:受虐属性
IF ABL:受虐属性
	LOCAL += MIN(ABL:受虐属性, 5) * 5
	LOCALS = %LOCALS% + 受虐属性LV{ABL:受虐属性}({MIN(ABL:受虐属性, 5)*5})
ENDIF
;害羞
IF TALENT:害羞
	LOCAL -= 15
	LOCALS = %LOCALS% - 害羞(15)
ENDIF
;不知羞恥
IF TALENT:不知羞恥
	LOCAL += 5
	LOCALS = %LOCALS% + 不知羞恥(5)
ENDIF
;倒錯的
IF TALENT:倒錯的
	LOCAL += 10
	LOCALS = %LOCALS% + 倒錯的(10)
ENDIF
IF TALENT:贞操重视
	LOCAL -= 20
	LOCALS = %LOCALS% - 贞操重视(20)
ENDIF

;合計を表示(80以上で実行)
CALL PRINT_COM_ORDER_COM, LOCALS, ARG, LOCAL, 80

RETURN RESULT


code:
@COM149
;导管
CALL PRINT_TRAIN_NAME(SELECTCOM)

;-------------------------------------------------
;実行できるかの判定
;-------------------------------------------------
CALL SHOW_COM_ORDER_COM, 149

;実行できない
SIF TCVAR:干劲 == 0
	RETURN 0
;-------------------------------------------------
;実行決定
;-------------------------------------------------
CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

RETURN 1

@FLAG_COM149, ARG
;尿的污垢発生
IF PENIS(TARGET)
	STAIN:Ｐ = SET_STAIN(STAIN:Ｐ, "尿")
ELSE
	STAIN:Ｖ = SET_STAIN(STAIN:Ｖ, "尿")
ENDIF
;利尿剤の効果は消失
TEQUIP:利尿剤 = 0
;主动放尿したので、最後まで出し切ります
BASE:尿意 = 0
;胖次をずらす
SIF !CHECK_CLO("Ｖ露出") && INSERT("挿入") == 0
	CALL SETFLAG, "ボトムを露出：テ接吻ト没有", TARGET
;放尿
EXP:放尿経験 += 1
NOWEX:放尿 += 1
EX:放尿 += 1

@SOURCE_COM149, ARG
CALL VARSET_COMF

CALL SOURCE_DOWNBASE_HP, 149

SOURCE:0:疼痛 = 1000
SOURCE:0:恐惧 = 500
SOURCE:0:不潔 = 500
SOURCE:0:露出 = 2000
SOURCE:0:逸脱 = 1500
SOURCE:0:屈従 = 3000

;PALAM:欲情をみる
SELECTCASE PALAM:欲情
CASE IS < PALAMLV:1
	TIMES SOURCE:0:逸脱, 1.50
CASE IS < PALAMLV:2
	TIMES SOURCE:0:逸脱, 1.00
CASE IS < PALAMLV:3
	TIMES SOURCE:0:逸脱, 0.90
CASE IS < PALAMLV:4
	TIMES SOURCE:0:逸脱, 0.70
CASE IS >= PALAMLV:4
	TIMES SOURCE:0:逸脱, 0.50
ENDSELECT

;ABL:受虐属性をみる
SELECTCASE ABL:受虐属性
CASE 0
	TIMES SOURCE:0:情愛, 0.80
	TIMES SOURCE:0:露出, 0.80
CASE 1
	TIMES SOURCE:0:情愛, 1.00
	TIMES SOURCE:0:露出, 1.00
CASE 2
	TIMES SOURCE:0:情愛, 1.30
	TIMES SOURCE:0:露出, 1.20
CASE 3
	TIMES SOURCE:0:情愛, 1.40
	TIMES SOURCE:0:露出, 1.40
CASE 4
	TIMES SOURCE:0:情愛, 1.70
	TIMES SOURCE:0:露出, 1.50
CASEELSE
	TIMES SOURCE:0:情愛, 2.00
	TIMES SOURCE:0:露出, 1.70
ENDSELECT

;場所をみる
IF TEQUIP:室内露出PLAY || TEQUIP:野外PLAY
	TIMES SOURCE:0:逸脱, 2.50
	TIMES SOURCE:0:露出, 1.50
ELSEIF TEQUIP:浴室PLAY
	TIMES SOURCE:0:逸脱, 0.50
ENDIF

SELECTCASE EXP:放尿経験 + EXP:漏尿経験
CASE IS < EXPLV:1
	TIMES SOURCE:0:逸脱, 3.00
CASE IS < EXPLV:2
	TIMES SOURCE:0:逸脱, 2.50
CASE IS < EXPLV:3
	TIMES SOURCE:0:逸脱, 2.00
CASE IS < EXPLV:4
	TIMES SOURCE:0:逸脱, 1.00
CASE IS < EXPLV:5
	TIMES SOURCE:0:逸脱, 0.80
CASEELSE
	TIMES SOURCE:0:逸脱, 0.60
ENDSELECT

SELECTCASE EXP:尿道経験 + MIN(EXP:PLAYER:尿道責め経験, 50)
CASE IS < 5
CASE IS < 20
	TIMES SOURCE:0:疼痛, 0.70
CASE IS < 50
	TIMES SOURCE:0:疼痛, 0.50
CASE IS < 150
	TIMES SOURCE:0:疼痛, 0.20
CASE IS < 300
	TIMES SOURCE:0:疼痛, 0.05
CASEELSE
	TIMES SOURCE:0:疼痛, 0.00
ENDSELECT

IF TALENT:尿道性感
	TIMES SOURCE:0:疼痛, 0.50
	TIMES SOURCE:0:恐惧, 0.50
ENDIF

;筋弛緩剤
SIF GETBIT(TEQUIP:筋弛緩剤, 1)
	TIMES SOURCE:0:疼痛, 0.20

CALL CALC, "調教ソース", ARG

@TRAIN_MESSAGE_COM149
INVERTBIT TEQUIP:导管, 0
IF TEQUIP:导管
	IF PENIS(TARGET)
		PRINTFORMW 将%CALLNAME:TARGET%的陰茎緊緊固定住後、用指尖微微拡寬尿道口、将塗上潤滑用的果凍的導管慢慢的挿入了……
		PRINTFORML …
		PRINTFORML ……
		PRINTFORML ………
		PRINTFORMW 大約、挿入了20cm長度的導管？
	ELSE
		PRINTFORMW 将%CALLNAME:TARGET%飽満的陰唇撥開、将露出的尿道口微微地拓寬後、将塗上潤滑用的果凍的導管慢慢的挿入了……
	ENDIF
	PRINTFORMW 導管的頂端被推到膀胱的位置、因挿入而繃緊的尿道稍稍放鬆後、金黄的尿液就順著導管淳淳流淌而出……
ELSE
	PRINTFORMW 従%CALLNAME:TARGET%的尿道中慎重小心的抜出了導尿管……
ENDIF

@EQUIP_COM149
SIF TEQUIP:导管 == 0
	RETURN 0

CALL SOURCE_COM149, 10
RETURN 1

@EQUIP_COM149_2
SIF TEQUIP:导管 == 0
	RETURN 0

PRINTL ＜导管挿入中＞
IF BASE:尿意 >= 1000 || (BASE:尿意 >= 100 && RAND:4 == 0)
	IF INSERT("挿入")
		SELECTCASE CONDS("体位")
		CASE "後背位"
			PRINTFORML 插入尿道的導管中、快樂的淫液混雜著尿液一併涓涓流下……
		CASEELSE
			PRINTFORML 插入尿道的導管中、尿液被斷斷續續湧出的愛液推出……
		ENDSELECT
	ELSE
		PRINTFORML 插入尿道的導管中、快樂的淫液混雜著尿液一併涓涓流下……
	ENDIF
	;尿的污垢発生
	IF PENIS(TARGET)
		STAIN:Ｐ = SET_STAIN(STAIN:Ｐ, "尿")
	ELSE
		STAIN:Ｖ = SET_STAIN(STAIN:Ｖ, "尿")
	ENDIF
	BASE:尿意 = 0
ENDIF

RETURN 1

@COM_ORDER_COM149, ARG
VARSET LOCAL
LOCALS =

;すべての命令に共通の要素を考慮
;(信頼が高いと命令に従いやすいなど)
CALL COM_ORDER, ARG

LOCAL = RESULT

;ABL:露出癖
IF ABL:露出癖
	LOCAL += MIN(ABL:露出癖, 10) * 2
	LOCALS = %LOCALS% + 露出癖LV{ABL:露出癖}({MIN(ABL:露出癖, 10) * 2})
ENDIF
;ABL:受虐属性
IF ABL:受虐属性
	LOCAL += MIN(ABL:受虐属性, 5) * 3
	LOCALS = %LOCALS% + 受虐属性LV{ABL:受虐属性}({MIN(ABL:受虐属性, 5) * 3})
ENDIF
;冷静
IF TALENT:冷静
	LOCAL -= 5
	LOCALS = %LOCALS% - 冷静(5)
ENDIF
;害羞
IF TALENT:害羞
	LOCAL -= 15
	LOCALS = %LOCALS% - 害羞(15)
ENDIF
;不知羞恥
IF TALENT:不知羞恥
	LOCAL += 2
	LOCALS = %LOCALS% + 不知羞恥(2)
ENDIF
;清楚
IF TALENT:清楚
	LOCAL -= 10
	LOCALS = %LOCALS% - 清楚(10)
ENDIF
;接受快感
IF TALENT:接受快感
	LOCAL += 5
	LOCALS = %LOCALS% + 接受快感(5)
ENDIF
;否定快感
IF TALENT:否定快感
	LOCAL -= 5
	LOCALS = %LOCALS% - 否定快感(5)
ENDIF
;倒錯的
IF TALENT:倒錯的
	LOCAL += 10
	LOCALS = %LOCALS% + 倒錯的(10)
ENDIF
IF TEQUIP:拘束
	LOCAL += 30
	LOCALS = %LOCALS% + 拘束中(30)
ENDIF
;媚薬
IF TEQUIP:媚薬
	LOCAL += 2
	LOCALS = %LOCALS% + 媚薬(2)
ENDIF
;尿道性感
IF TALENT:尿道性感
	LOCAL += 10
	LOCALS = %LOCALS% + 尿道性感(10)
ELSE
	LOCAL -= 30
	LOCALS = %LOCALS% - 尿道性感没有(30)
ENDIF
;视频拍摄
IF TEQUIP:攝像機
	LOCAL -= 20
	LOCALS = %LOCALS% - 撮影中(20)
ENDIF
;野外or室内露出
IF TEQUIP:野外PLAY
	LOCAL -= 10
	LOCALS = %LOCALS% - 野外PLAY(10)
ELSEIF TEQUIP:室内露出PLAY
	LOCAL -= 10
	LOCALS = %LOCALS% - 室内露出(10)
ENDIF

;合計を表示(70以上で実行)
CALL PRINT_COM_ORDER_COM, LOCALS, ARG, LOCAL, 70

RETURN RESULT

license: unknown


code:
@COM14
;扒開菊穴　Ｖ挿入の時には体勢的に可能ならば抜かずに行う
CALL PRINT_TRAIN_NAME(SELECTCOM)

;-------------------------------------------------
;実行できるかの判定
;-------------------------------------------------
CALL SHOW_COM_ORDER_COM, 14

;実行できない
SIF TCVAR:干劲 == 0
	RETURN 0
;-------------------------------------------------
;実行決定
;-------------------------------------------------
CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

RETURN 1

@FLAG_COM14, ARG
;腸内ルートは角色の指⇔Ａ的污垢が移動
SIF CHECK_COM("腸内描写", 14)
	CALL EQ_STAIN, TARGET, "手", TARGET, "Ａ"
SIF PREVCOM == 13 && CHECK_COM("両穴くぱぁ")
	CALL FLAG_COM13, 50

@SOURCE_COM14, ARG
SIF PREVCOM == 13 && CHECK_COM("両穴くぱぁ")
	CALL SOURCE_COM13, 50

CALL VARSET_COMF

CALL SOURCE_DOWNBASE_HP, 14

SOURCE:0:不潔 = 500
SOURCE:0:露出 = 4000
SOURCE:0:屈従 = 1000
SOURCE:0:逸脱 = 1000

;ABL:露出癖をみる
SELECTCASE ABL:露出癖
CASE 0 TO 5
	SOURCE:0:充足 = ABL:露出癖*500
CASE 6 TO 10
	SOURCE:0:充足 = 2500 + (ABL:露出癖 - 5)*200
CASEELSE
	SOURCE:0:充足 = 3500 + (ABL:露出癖 - 10)*100
ENDSELECT

;害羞・不知羞恥
SIF TALENT:害羞
	TIMES SOURCE:0:逸脱, 2.00
SIF TALENT:不知羞恥
	TIMES SOURCE:0:逸脱, 0.50

IF TALENT:Ａ性向
	TIMES SOURCE:0:不潔, 0.50
	TIMES SOURCE:0:充足, 2.00
	TIMES SOURCE:0:逸脱, 0.50
ENDIF

;视频拍摄、室内露出、野外PLAYは経験等にプラス
IF TEQUIP:攝像機
	TIMES SOURCE:0:露出, 1.50
	TIMES SOURCE:0:屈従, 1.50
	TIMES SOURCE:0:逸脱, 1.50
ENDIF
IF TEQUIP:室内露出PLAY || TEQUIP:野外PLAY
	TIMES SOURCE:0:露出, 1.50
	TIMES SOURCE:0:屈従, 1.50
	TIMES SOURCE:0:逸脱, 1.50
ENDIF
;恥ずかしさアップなPLAY
IF PREVEX("絶頂数合計") || PREVEX("Ａ中出し")
	TIMES SOURCE:0:露出, 1.50
	TIMES SOURCE:0:逸脱, 1.50
ENDIF

CALL CALC, "調教ソース", ARG

@TRAIN_MESSAGE_COM14
#DIM FLAG_MESSAGE, 2

;ルートとしては、腸の中まで見せるor菊花の皺を伸ばすor尻肉を広げて肛門を見せる
IF PREVCOM == SELECTCOM
	IF COND("両穴くぱぁ")
		PRINTFORMW %CALLNAME:TARGET%面向%CALLNAME:PLAYER%、保持著將股間兩個汁水淋漓的淫穴大大張開展示著的姿勢、一動不動……
	ELSE
		PRINTFORMW %CALLNAME:TARGET%將朝向%CALLNAME:PLAYER%並將肛門推開的姿勢保持著、一動不動……
	ENDIF
	RETURN 0
ENDIF

;両穴くぱぁ
IF PREVCOM == 13 && CHECK_COM("両穴くぱぁ")
	IF COND("両穴くぱぁ")
		PRINTFORM %CALLNAME:TARGET%已經將兩個淫穴毫不憐惜的張開了、但還是再次響應了要求、
	ELSE
		PRINTFORM %CALLNAME:TARGET%在將尻穴明顯的展示著的同時、更加突出腰肢、
	ENDIF
;前回Ｖくぱぁ
ELSEIF PREVCOM == 13
	PRINTFORM %CALLNAME:TARGET%保持著將陰部清楚地展示出来的姿勢、這次將
ELSE
	;脚拘束
	IF GETBIT(TEQUIP:拘束, 21)
		PRINTFORM 兩腳被拘束具大大地拉扯開的%CALLNAME:TARGET%、
	ELSE
		PRINTFORM %CALLNAME:TARGET%
	ENDIF

	;穴開きでない胖次を是ていて、ずらされていないなら、必ずずらす
	SIF !CHECK_CLO("Ａ露出")
		CALL SETFLAG, "ボトムを露出：テ接吻ト没有", TARGET

	VARSET FLAG_MESSAGE
	;表情、雰囲気描写
	IF TCVAR:干劲 <= 20
		;害羞
		IF TALENT:害羞
			PRINTFORM %TEXTR("緊緊閉上雙瞳/低下頭/羞澀的合上雙眼/羞恥的臉頰通紅/羞恥的臉頰通紅")%
			FLAG_MESSAGE = 1
		;叛逆
		ELSEIF TALENT:叛逆
			PRINTFORM 滿臉嫌惡而輕蔑
			FLAG_MESSAGE = 1
		ENDIF
	ELSEIF TALENT:恋慕 == 0
		;害羞
		IF TALENT:害羞
			PRINTFORM %TEXTR("羞恥的臉頰通紅/臉上就像發燒了一樣/稍稍的染紅臉頰/臉頰浮現出薄薄紅暈")%
			FLAG_MESSAGE = 1
		ENDIF
	ELSEIF TALENT:恋慕
		;害羞
		IF TALENT:害羞 && RAND:2 == 0
			PRINTFORM 羞恥浮現在臉頰上
		ELSEIF TEQUIP:六九式 == 0 && RAND:3 == 0
			PRINTFORM 視線偷偷向上飄、窺視著情況
		ELSEIF RAND:2 == 0
			PRINTFORM 臉頰浮現出薄薄緋紅
		ELSE
			PRINTFORM 臉頰發燙
		ENDIF
		FLAG_MESSAGE = 1
	ENDIF
	IF TEQUIP:六九式
		PRINTFORM 、向%CALLNAME:PLAYER%的臉頰挺起腰肢
		IF TCVAR:干劲 > 20 && RAND:2
			PRINT 、同時將雙腿張的更開、將
		ELSE
			PRINT 、隨後將
		ENDIF
	;脚を拘束されていないなら開く描写
	ELSEIF COND("脚の使用") && INSERT("Ｖ挿入") == 0
		;まんぐり返し
		IF TCVAR:干劲 > 50 && RAND:2
			IF RAND:2
				PRINTFORM 、保持著將小穴向上舉起的仰躺的姿勢、
			ELSE
				PRINTFORM 、將雙腿向上方翻起、
			ENDIF
			IF RAND:2 == 0
				PRINTFORM 、用雙手將大腿抓住並將股間拉得更開、將
			ELSEIF TEQUIP:攝像機
				PRINTFORM 、為了能被拍攝的更加明晰而將腰肢舉起、將
			ELSE
				PRINTFORM 、為了能讓%CALLNAME:PLAYER%清楚的看見而突出腰部、將
			ENDIF
			IF ABL:露出癖 >= 3 && CHOSECOM(SELECTCOM) == 0 && RAND:2
				IF RAND:2
					PRINTFORM 、主動將視線誘導集中在股間的同時、將
				ELSE
					PRINTFORM 、享受著集中在股間上的灼熱視線的同時、將
				ENDIF
			ELSE
				PRINTFORM 、將
			ENDIF
		;お尻を向ける
		ELSE
			IF TALENT:不知羞恥 && TCVAR:干劲 > 20 && RAND:2
				PRINTFORM 毫不猶豫地
			ELSEIF (TALENT:内気 || TALENT:坦率 || TALENT:害羞) && RAND:2
				PRINTFORM 怯生生地
			ENDIF
			SIF TCVAR:干劲 > 20 && RAND:2 == 0
				PRINTFORM 做出四肢著地的姿勢、
			PRINTFORM 將尻部高高舉起、將
		ENDIF
	ELSEIF FLAG_MESSAGE
		PRINTFORM 、
	ENDIF
ENDIF

IF TCVAR:干劲 > 20 && RAND:2
	LOCALS = 更加地
ELSEIF RAND:2
	LOCALS = 緩慢地
ELSE
	LOCALS = 輕柔地
ENDIF

;腸内
IF TCVAR:干劲 > 20 && ABL:Ａ拡張 >= 2 && COND("指の使用") && COND("液体浣腸") + COND("排泄：危険域") == 0 && (PREVCOM == 142 || PREVCOM == 143 || RAND:3 == 0)
	IF PREVCOM == 142 || PREVCOM == 143
		IF PREVPLAY("結腸責め")
			PRINTFORM %TEXTR("剛剛被手腕侵犯了結腸/不久前被緊握的拳頭開闢了結腸/將腸道深處翻攪的拳頭拔出")%的
		ELSE
			PRINTFORM 剛剛被手腕來回翻攪過的
		ENDIF
	ELSEIF PREVCOM == 33
		PRINTFORM 剛剛拔出振動棒的
	ELSEIF PREVCOM >= 70 && PREVCOM <= 79
		PRINTFORM 剛剛被猛烈侵犯的
	ELSEIF PREVEX("Ａ絶頂")
		IF PREVCOM == 3
			PRINTFORM 剛剛迎來高潮的
		ELSE
			PRINTFORM 刚刚被欺负的
		ENDIF
	ELSEIF EX:Ａ絶頂
		PRINTFORM 在极致快感下溶化的
	ENDIF

	IF EXP_GAIN("Ａ拡張経験") >= 3 || ABL:Ａ拡張 >= 3 + RAND:3 || PREVCOM == 142 || PREVCOM == 143
		PRINTFORM 松弛了
	ELSEIF RAND:2 == 0
		PRINTFORM 依然保有正常色澤的
	ENDIF

	;両穴くぱぁ且つ継続中ではない
	IF PREVCOM == 13 && CHECK_COM("両穴くぱぁ") && COND("両穴くぱぁ") == 0
		PRINTFORM 菊穴にも指を潜り込ませて
	ELSE
		PRINTFORM 將手指吞沒的尻穴
	ENDIF
	IF RAND:3 == 0
		PRINTFORMW 、朝向%CALLNAME:PLAYER%的方向%LOCALS%張開了……
	ELSEIF RAND:2 == 0
		PRINTFORMW %LOCALS%展開了……
	ELSE
		PRINTFORMW %LOCALS%打開、被清楚地觀賞著……
	ENDIF
	FLAG_MESSAGE:1 = 1
;菊花
ELSE
	;両穴くぱぁ継続
	IF COND("両穴くぱぁ")
		PRINTFORM 尻肉一併推開、
	ELSE
		PRINTFORM %TEXTS("お尻：視覚")%的柔肉%LOCALS%推展開、
	ENDIF

	LOCALS =
	SIF TALENT:好色 + TALENT:淫乱 && TALENT:尻穴狂い && PALAM:欲情 >= 3000 && RAND:2
		LOCALS = 淫亂的

	IF RAND:2 && TALENT:肤色
		SELECTCASE CONDS("肤色")
		CASE "褐色肌"
			PRINTFORM 與健康的褐色肌膚相比、也顯得格外醒目的可可茶色的%LOCALS%
		CASE "色白"
			PRINTFORM 與晶莹剔透的白暫肌膚一致、顏色稍淺的粉嫩的%LOCALS%
		CASE "青肌"
			PRINTFORM 與異於常人的青藍肌膚相同、色調更加深沉而更顯妖冶的%LOCALS%
		ENDSELECT
	;陰毛描写
	ELSEIF COND("陰毛") >= 6 && AGE(TARGET) >= 3 && RAND:2
		PRINTFORM 被凌亂的陰毛装飾著邊緣的%LOCALS%
	ELSEIF COND("液体浣腸") + COND("排泄：危険域")
		PRINTFORM 將排泄的刺激\@ COND("排泄：限界") ? 以幾乎馬上就要一瀉千里的姿態勉強的忍受著的 # 拼命忍耐著的 \@%LOCALS%
	ELSEIF PREVCOM >= 70 && PREVCOM <= 79
		PRINTFORM 剛剛被猛烈侵犯的%LOCALS%
	ELSEIF ABL:Ａ感覚 >= 3 && RAND:2
		PRINTFORM 因興奮而充血並不斷顫抖著的%LOCALS%
	ELSE
		IF ABL:Ａ拡張 >= 4 && EXP:Ａ経験 >= 150
			PRINTFORM 被充分開發的%LOCALS%
		ELSEIF ABL:Ａ拡張 >= 2 && EXP:Ａ経験 >= 50
			PRINTFORM 十分柔軟的%LOCALS%
		ELSEIF AGE(TARGET) == 1 && RAND:2
			PRINTFORM 小小的
		ELSEIF RAND:2
			PRINTFORM 可愛的
		ELSE
			PRINTFORM 依然還十分緊致的
		ENDIF
	ENDIF

	IF RAND:3 == 0
		PRINTFORM 菊穴
	ELSEIF RAND:2 == 0
		PRINTFORM 尻穴
	ELSE
		PRINTFORM 肛門
	ENDIF
	;両穴くぱぁ且つ継続中ではない
	IF PREVCOM == 13 && CHECK_COM("両穴くぱぁ") && COND("両穴くぱぁ") == 0
		PRINTFORM も
	ELSE
		PRINTFORM を
	ENDIF
	IF TCVAR:干劲 > 50 && RAND:2
		PRINTFORMW 清楚地展示著……
	ELSEIF RAND:2
		PRINTFORMW 暴露著內部……
	ELSE
		PRINTFORMW 剝開、完全被一覽無遺……
	ENDIF
ENDIF

SIF TEXTS("Ａコンディション", TARGET) != ""
	CALL PRINT_STRW, @"(%CALLNAME:TARGET%的Ａ状态为_黄色_%TEXTS("Ａコンディション", TARGET)%_)"

;前回Ａ性交
IF PREVCOM >= 70 && PREVCOM <= 79 && COND("液体浣腸") + COND("排泄：危険域") == 0
	IF ABL:Ａ拡張 >= 3 && RAND:3 == 0
		IF EX:Ａ中出し >= 3 && RAND:3 == 0
			PRINTFORMW 從渾圓的張開的尻穴中被粗暴注入的各種液體與內部的體液混合起來、隨著填充物的拔出而淒慘的衝出體內。隨著令人羞恥的臭味與巨響、渾濁而骯髒的液體飛濺而出……
		ELSE
			PRINTFORM 誇張地展開著的肛門、隨著在邊緣不斷翻攪的手指而讓腸道中渾濁的污液不斷泛起白沫、
			SIF PREVEX("Ａ中出し")
				PRINT 裹挟着被注入体内的灼热白浊一起、
			PRINTFORMW 黏黏糊糊地垂落下来…
		ENDIF
	;前回中出し
	ELSEIF PREVEX("Ａ中出し")
		IF TCVAR:干劲 > 50 && RAND:2
			PRINTFORMW %CALLNAME:TARGET%稍稍收緊小腹、就像是在確認之前快樂性愛的結果一樣、讓精液緩緩逆流而出……
		ELSEIF RAND:2 == 0
			PRINTFORMW 被满满地中出的小穴、因为快感而张開着阴道口、沾满了精液的粘膜清晰可见…
		ELSE
			PRINTFORMW 正配合著凌亂的氣息而嫵媚的顫抖著的膣口之中、剛剛灌注進去的精液隨著肛門口的抖動而灑落……
		ENDIF
	ELSEIF ABL:Ａ拡張 >= 2 && RAND:2 == 0
		PRINTFORMW 在無法閉合的媚肉洞窟的最深處、淫靡的粘膜正因尋求失去的灼熱快感而渴望的蠢動著……
	ELSE
		PRINTFORMW 陰莖拔出後的肛門、因消失的灼熱填充物與快感而淫靡地顫抖著、任由\@ RAND:2 ? 柔軟淫靡的 # 色調鮮明的 \@內部粘膜暴露出來……
	ENDIF
	TFLAG:地の文分岐 = 1
;腸内ルート
ELSEIF FLAG_MESSAGE:1 == 1
	TFLAG:地の文分岐 = 2
	;拡張度大
	IF EXP_GAIN("Ａ拡張経験") >= 5 || ABL:Ａ拡張 >= 3 + RAND:3
		IF RAND:3 == 0
			IF CHOSECOM(SELECTCOM) == 0 && BASE:排泄物 == 0 && EX:排泄
				PRINTFORM 內部的髒污被完全排空的
			ELSE
				SIF TALENT:処女 == 1 && CHOSECOM(SELECTCOM) == 0
					PRINTFORM 與從未使用的純潔陰部形成鮮明對照的
				IF CHOSECOM(SELECTCOM) == 0 && ABL:Ａ技術 < ABL:Ａ拡張
					PRINTFORM 令人對\@ AGE(TARGET) < 3 ? 將來發育 # 日常生活 \@產生擔心的\@ RAND:2 ? 格外松弛的 # 大張着的 \@
				ELSEIF RAND:2 == 0
					PRINT 正清楚地大開的
				ELSE
					PRINT 每一處褶皺都伸展開來的
				ENDIF
			ENDIF
			PRINTFORM 尻穴、
			IF EXP:Ａ性交経験 && EXP:Ａ拳交経験 && RAND:2
				PRINTFORM 將接受過各種各樣物品侵犯的
			ELSE
				PRINTFORM 在將羞恥的%TEXTS("气味")%不斷吐出的同時、將
			ENDIF
			PRINTFORMW 直腸內部的景色毫無遮掩的向%CALLNAME:PLAYER%暴露著……
			TFLAG:地の文分岐 = 3
		ELSEIF RAND:2 == 0
			IF ABL:Ａ感覚 >= 3 && EXP:Ａ経験 >= 150 && EXP:Ａ性交経験 >= 50 && RAND:2
				PRINTFORM 早已被視作性器開發完全的排泄孔
			ELSE
				PRINTFORM 正清楚地打開的肛門
			ENDIF
			IF INSERT("Ｖ挿入")
				PRINTFORMW 正與在前穴中不斷衝擊的\@ PENIS(PLAYER) ? 陰莖 # 道具 \@配合的蠢動著、將腸道中的%TEXTS("气味")%毫不介意的散播開來……
			ELSEIF RAND:2
				PRINTFORMW 正配合著凌亂的氣息而嫵媚的顫抖著、就好像是在懇求更進一步的玩弄而自發地展示著……
			ELSE
				PRINTFORMW 正配合著凌亂的氣息而嫵媚的顫抖著、妖媚的引誘著%CALLNAME:PLAYER%……
			ENDIF
		ELSE
			PRINTFORM 被粗暴地蹂躪著的肛門、
			SIF EXP_GAIN("Ａ拡張経験") >= 3 || EXP_GAIN("Ａ性交経験") >= 5 || PREVCOM == 142 || PREVCOM == 143
				PRINTFORM 即使失去了原有的堵塞物也仍然
			PRINTFORM 誇張的張開著、
			SIF PREVEX("Ａ中出し") && CHECK_STAIN(STAIN:Ａ, "精液")
				PRINT 混雜著灼熱白濁的
			PRINTFORMW 腥臭的污液與腸液一同飛濺而出……
		ENDIF
	ELSEIF EX:Ａ絶頂 >= 5
		PRINTFORM %CALLNAME:TARGET%的陰部
		SIF INSERT("Ｖ挿入")
			PRINTFORM 在膣道的最深處隨著刺入的陰莖而糾纏起來的同時、
		PRINTFORM 無比貪婪地
		IF EXP_GAIN("肛内精液経験")
			PRINTFORMW 將腸道中腥臭渾濁的污液噴灑著……
		ELSE
			PRINTFORMW 抽搐並反復的收縮著……
		ENDIF
	ELSEIF PALAM:欲情 >= 3000
		PRINTFORM 形成了淫亂洞窟的媚肉與直腸的粘膜、
		IF INSERT("Ｖ挿入")
			PRINTFORM 像是對膣道傳遞的快感刺激做出反應一樣、
		ELSE
			PRINTFORM 即使沒有受到任何觸碰、也會
		ENDIF
		PRINTFORMW 淫亂的抽搐痙攣著……
	ENDIF
;菊花ルート
ELSE
	;前回中出し
	IF PREVEX("Ａ中出し")
		IF COND("液体浣腸") + COND("排泄：危険域")
			PRINTFORMW %CALLNAME:TARGET%將剛剛被侵犯過的肛門用力緊緊地收緊、將混雜著精液的腸內污物\@ ABL:受虐属性 < 3 ? 拼命地 # \@壓回體內、忍耐住了……
		ELSEIF TCVAR:干劲 > 50 && RAND:2
			PRINTFORMW %CALLNAME:TARGET%稍稍收緊小腹、就像是在確認之前快樂性愛的結果一樣、讓精液緩緩逆流而出……
		ELSE
			PRINTFORMW 正配合著凌亂的氣息而嫵媚的顫抖著的腸道之中、剛剛灌注進去的精液隨著肛門的抖動而灑落……
		ENDIF
	ELSEIF TALENT:好色 + TALENT:淫乱 && ABL:Ａ感覚 >= 3 && ABL:露出癖 >= 3 && ABL:Ａ技術 >= 5 && ABL:Ａ拡張 >= 2 && PREVEX("Ａ絶頂") == 0 && !RAND:3
		IF COND("液体浣腸") + COND("排泄：危険域") || (PALAM:欲情 >= 3000 && RAND:2)
			PRINTFORM %CALLNAME:TARGET%\@ COND("指の使用") ? 用指尖 # \@將不断抽搐着的肛門慢慢拉開、同時
			IF TALENT:強気 && RAND:2
				PRINT 大膽的
			ELSE
				PRINT 妖艶
			ENDIF
			PRINTFORMW 笑容自然地浮現在臉上……
		ELSE
			PRINTFORMW %CALLNAME:TARGET%將緊緊地收攏著的緊窄肛門、以挑逗性的姿勢張開來、誘惑著%CALLNAME:PLAYER%深入其中……
		ENDIF
	ELSEIF PALAM:欲情 >= 10000 && COND("Ａ熟達")
		IF TALENT:害羞 && RAND:2
			SIF INSERT("Ｖ挿入")
				PRINTFORM 阴道道正在被不斷凌辱的
			PRINTFORM 臉頰通紅的%CALLNAME:TARGET%
			IF TALENT:清楚 && RAND:3
				PRINTFORM 仍然保持著清純可憐的氣質、
			ELSEIF TALENT:内気 && RAND:3
				PRINTFORM 仍然顯得柔弱可欺的樣子、
			ELSE
				PRINTFORM 顯得非常羞恥的樣子、
			ENDIF
			PRINTFORM 而與之相反的、
			IF ABL:Ａ拡張 >= 4 && EXP_GAIN("Ａ拡張経験")
				PRINTFORMW 貪婪的尻穴正在大張著洞口、以嫵媚的姿態發出邀請……
			ELSE
				PRINTFORMW 充血興奮的尻穴顯得十分飢渴、以嫵媚的姿態發出邀請……
			ENDIF
		ELSEIF ABL:Ａ拡張 >= 3 && RAND:2
			SIF INSERT("Ｖ挿入")
				PRINTFORM 因膣道被強烈的凌辱著而更加興奮的
			PRINTFORM %CALLNAME:TARGET%的肛門\@ EX:絶頂数合計 ? 完全的 # 已經 \@充血綻開、
			IF COND("温泉浣腸")
				PRINTFORMW 仍由灌入的尿液慢慢地滲出…
			ELSEIF COND("精液浣腸")
				PRINTFORMW 任由渾濁的白色灼液慢慢滲出……
			ELSEIF COND("液体浣腸")
				PRINTFORMW 任由浣腸液與腸內的污液一併慢慢滲出……
			ELSE
				PRINTFORMW 正淫靡地張合的肛門、正像是在邀請一樣隨著呼吸顫抖著……
			ENDIF
		ELSE
			SIF INSERT("Ｖ挿入")
				PRINTFORM %CALLNAME:PLAYER%在子宮的入口隔著一層薄薄的肉襞將腸道摩擦著、
			PRINTFORMW 凝視著近在眼前的、舒展開褶皺的肛門、\@ COND("液体浣腸") + COND("排泄：危険域") ? 忍耐著內外的刺激以阻止內部渾濁液的漏出、 # 因羞耻与对欲望的渴求而不断颤抖着、 \@随着收缩而产生的阴影显得格外诱人……
		ENDIF
	ELSEIF PALAM:欲情 >= 3000 || TALENT:恋慕 || INSERT("Ｖ挿入")
		IF ABL:Ａ拡張 >= 3 && COND("液体浣腸") + COND("排泄：危険域") == 0 && RAND:3 == 0
			PRINTFORM %CALLNAME:TARGET%那擴張完畢的肛門、
			IF INSERT("Ｖ挿入")
				PRINTFORMW 與膣道週邊的媚肉相互拉扯著、讓前後兩張淫口隨著凌亂的喘息不住張合起來……
			ELSEIF COND("指の使用") == 0 || RAND:2
				PRINTFORMW 像是在焦慮的期待什麼一樣、不需要指尖的輔助、就能自行張開……
			ELSE
				PRINTFORMW 不需要手指的輔助就能自行一點點張開……
			ENDIF
		ELSEIF TALENT:恋慕 && RAND:3 == 0
			PRINTFORMW %CALLNAME:TARGET%的肛門因羞恥而被一層薄薄的汗液浸濕、以絕妙的艷麗氛圍讓%CALLNAME:PLAYER%的雙眼格外享受……
		ELSEIF PALAM:欲情 >= 6000
			IF INSERT("Ｖ挿入")
				PRINTFORM 因不斷出入膣道的\@ PENIS(PLAYER) ? 陰莖 # 道具 \@而扭曲著收緊的
			ELSE
				PRINTFORM 所有褶皺被更加的拉扯著展開的
			ENDIF
			PRINTFORM %CALLNAME:TARGET%的肛門、興奮的
			IF EXP:Ａ経験 >= 50 && RAND:2
				PRINTFORMW 變得更加興奮與熱烈、就像是在渴望被觸碰一樣誘人的顫抖著……
			ELSE
				PRINTFORMW 隨著肛門的顫抖而洋溢著汗液與腸液的混合物、同時獨特而令人上癮的劇烈%TEXTS("气味")%正漂浮在半空……
			ENDIF
		ELSE
			SIF INSERT("Ｖ挿入")
				PRINTFORM 在被激烈的侵犯著膣道的同時、
			PRINTFORMW 輕輕地隨著凌亂喘息抖動的嫵媚淫肉將%CALLNAME:PLAYER%的視線牢牢吸引住、久久無法離開……
		ENDIF
	ENDIF
	TFLAG:地の文分岐 = 4
ENDIF

;両穴くぱぁフラグＯＮ
SIF PREVCOM == 13 && CHECK_COM("両穴くぱぁ")
	CALL SETFLAG, "両穴くぱぁ", TARGET

CALL TRAIN_PHOTO_COM14

@TRAIN_PHOTO_COM14
;くぱぁ直後の写真撮影で使う
STR:青写真 =

SIF INSERT("Ｖ挿入")
	STR:青写真 += @"%TEXTR("一边进行着膣穴性交/一边用前穴品尝着阴茎/一边露出因为性交而感到非常幸福的表情")%"

STR:青写真 += @"%TEXTS("お尻：感触")%被押開、"

SIF COND("両穴くぱぁ")
	STR:青写真 += @"%TEXTR("膣穴/膣口")%和"

IF CHECK_COM("腸内描写")
	IF PREVPLAY("排泄")
		STR:青写真 += @"排泄直後%TEXTR("干净的直腸粘膜/变得干净了的直腸")%"
	ELSEIF PREVPLAY("温泉浣腸")
		STR:青写真 += @"尿液流込直後的腸内"
	ELSEIF PREVEX("Ａ中出し")
		STR:青写真 += @"刚刚被排泄精液的直腸内"
	ELSE
		STR:青写真 += @"%TEXTR("大大広開的直腸内的景色/色泽鲜艳的直腸粘膜")%被"
	ENDIF
ELSE
	IF PREVEX("Ａ中出し")
		STR:青写真 += @"%TEXTR("中出性交直後的/粘液混合精液垂下的/中出直後精液溢出的")%肛門"
	ELSEIF PREVCOM >= 70 && PREVCOM <= 79
		STR:青写真 += @"%TEXTR("被使用之後的肛門/刚刚完成工作的排泄穴")%"
	ELSEIF PREVCOM == 142 || PREVCOM == 143
		STR:青写真 += @"拳交直後%TEXTR("大口張開着的肛門/大大広開的肛門/松弛開口的肛門")%"
	ELSE
		STR:青写真 += @"%TEXTS("肛門：視覚")%"
	ENDIF
ENDIF

STR:賞賛内容 = 将%STR:青写真%%TEXTR("露出来/無防備地暴露出来/暴露出来")%

IF TCVAR:干劲 > 20 && PENIS(TARGET) && COND("要去了") && CHECK_CLO("Ｐ露出")
	SELECTCASE SIZE("阴茎", TARGET)
	CASE 0
		STR:青写真 += @"和一起勃起至坚硬的%TEXTR("粗鸡鸡/兒童金金/短小阴茎")%"
	CASE 1
		IF COND("真性包茎")
			STR:青写真 += @"和一起勃起至坚硬的包茎阴茎"
		ELSEIF COND("假性包茎")
			STR:青写真 += @"和一起勃起至坚硬的阴茎"
		ELSE
			STR:青写真 += @"和一起勃起至坚硬的露茎阴茎"
		ENDIF
	CASE 2 TO 4
		STR:青写真 += @"和一起勃起至坚硬的巨根"
	CASE IS >= 5
		STR:青写真 += @"和一起勃起至坚硬的規格外の巨根"
	ENDSELECT
ENDIF

IF INSERT("Ｖ挿入")
	STR:青写真 += "露在了空气中"
ELSE
	STR:青写真 += @"被%PHOTO_SHOWING("見せつけ")%"
ENDIF

@COM_ORDER_COM14, ARG
VARSET LOCAL
LOCALS =

;すべての命令に共通の要素を考慮
;(信頼が高いと命令に従いやすいなど)
CALL COM_ORDER, ARG

LOCAL = RESULT

;ABL:露出癖
IF ABL:露出癖
	LOCAL += MIN(ABL:露出癖, 10) * 2
	LOCALS = %LOCALS% + 露出癖LV{ABL:露出癖}({MIN(ABL:露出癖, 10) * 2})
ENDIF
;ABL:受虐属性
IF ABL:受虐属性
	LOCAL += MIN(ABL:受虐属性, 5) * 3
	LOCALS = %LOCALS% + 受虐属性LV{ABL:受虐属性}({MIN(ABL:受虐属性, 5) * 3})
ENDIF
;冷静
IF TALENT:冷静
	LOCAL -= 5
	LOCALS = %LOCALS% - 冷静(5)
ENDIF
;害羞
IF TALENT:害羞
	LOCAL -= 25
	LOCALS = %LOCALS% - 害羞(25)
ENDIF
;不知羞恥
IF TALENT:不知羞恥
	LOCAL += 5
	LOCALS = %LOCALS% + 不知羞恥(5)
ENDIF
;清楚
IF TALENT:清楚
	LOCAL -= 20
	LOCALS = %LOCALS% - 清楚(20)
ENDIF
;否定快感
IF TALENT:否定快感
	LOCAL -= 5
	LOCALS = %LOCALS% - 否定快感(5)
ENDIF
;前回中出し
IF PREVEX("Ａ中出し")
	LOCAL -= 10
	LOCALS = %LOCALS% - 中出し直後(10)
ENDIF
IF COND("陶酔")
	LOCAL += 10
	LOCALS = %LOCALS% + 陶酔(10)
ENDIF
IF TEQUIP:媚薬
	LOCAL += 2
	LOCALS = %LOCALS% + 媚薬(2)
ENDIF
;视频拍摄
IF TEQUIP:攝像機
	LOCAL -= 30
	LOCALS = %LOCALS% - 撮影中(30)
ENDIF
;野外or室内露出
IF TEQUIP:野外PLAY
	LOCAL -= 10
	LOCALS = %LOCALS% - 野外PLAY(10)
ELSEIF TEQUIP:室内露出PLAY
	LOCAL -= 10
	LOCALS = %LOCALS% - 室内露出(10)
ENDIF

;合計を表示(40以上で実行)
CALL PRINT_COM_ORDER_COM, LOCALS, ARG, LOCAL, 40

RETURN RESULT


code:
@COM150
;深喉
CALL PRINT_TRAIN_NAME(SELECTCOM)

;-------------------------------------------------
;実行できるかの判定
;-------------------------------------------------
IF TEQUIP:深喉 == 0
	CALL SHOW_COM_ORDER_COM, 150
	;実行できない
	SIF TCVAR:干劲 == 0
		RETURN 0
ENDIF
;-------------------------------------------------
;実行決定
;-------------------------------------------------

CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

;-------------------------------------------------
;射精格子
;-------------------------------------------------
CALL SAMEN_CHECK

RETURN 1

@FLAG_COM150, ARG
CALL SETFLAG, "口交フラグ", TARGET
;対象の口⇔調教者のＰ的污垢が移動
CALL EQ_STAIN, TARGET, "Ｍ", PLAYER, "Ｐ"

@SOURCE_COM150, ARG
;口交と強制口交の合わせ技
IF TEQUIP:深喉
	CALL SOURCE_COM91, 30
	CALL SOURCE_COM140, 50
ELSE
	CALL SOURCE_COM91, 20
	CALL SOURCE_COM140, 20
ENDIF

@TRAIN_MESSAGE_COM150
;深喉解除(今のところは死に分岐)
IF TEQUIP:深喉
	;前回は強制口交
	IF PREVCOM == 140 && COND("前Ｔ調教者") == PLAYER && TEQUIP:推倒 == 0
		PRINTFORM %CALLNAME:PLAYER%將%CALLNAME:TARGET%的
		IF RAND:2
			PRINTFORM 口腔中衝刺的陰莖停下、挺直腰肢、
		ELSE
			;角が有ますか？
			PRINTFORM \@ COND("角ハンドル") ? 犄角緊緊抓住 # 頭顱輕輕按住 \@挺直腰肢、
		ENDIF
		;大きいと食道にまで行ってしまう
		IF SIZE("阴茎", PLAYER) >= 3
			PRINTFORMW 將深深侵入食道的陰莖慢慢地抽出……
		ELSE
			PRINTFORMW 將侵犯著口腔深處的陰莖慢慢地抽出……
		ENDIF
	ELSE
		PRINTFORMW %CALLNAME:TARGET%輕輕後仰、將直到根部也被完全嚥下的%CALLNAME:PLAYER%的陰莖解放出來……
	ENDIF
	TEQUIP:深喉 = 0
	RETURN 0
ENDIF

PRINTFORM %CALLNAME:TARGET%
;前回口交or強制口交
IF (PREVCOM == 91 || PREVCOM == 140) && COND("前Ｔ調教者") == PLAYER
	IF RAND:2 && CHECK("喉姦") >= 2
		IF RAND:3 == 0 && SIZE("阴茎", PLAYER) >= 3
			PRINTFORM 蹂躪著食道的
		ELSE
			PRINTFORM %TEXTR("毫無顧慮的使用著整個口腔的/在柔韌的舌頭上滑動著的")%
		ENDIF
	ELSEIF TEQUIP:推倒
		PRINTFORM 臉頰深深埋在股間的
	ELSEIF PREVCOM == 140 && RAND:2
		PRINTFORM 主動搖動腰肢來配合的
	ELSE
		PRINTFORM 不禁露出恍惚神情的
	ENDIF
	PRINTFORM %CALLNAME:PLAYER%の
;被逆推ているとき
ELSEIF TEQUIP:推倒
	;直前の調教と汚れで分岐
	IF COND("Ｐ汚れ", PLAYER)
		SIF (PREVCOM >= 70 && PREVCOM <= 79) && COND("前Ｔ調教者") == PLAYER && TEQUIP:眼罩 == 0
			PRINT 從自己的肛門剛剛拔出、
		;容易濕で追加
		SIF TALENT:容易濕 && RAND:3 && TEQUIP:眼罩 == 0
			PRINT べっとりと
		SIF (PREVCOM >= 60 && PREVCOM <= 69) && COND("前Ｔ調教者") == PLAYER && TEQUIP:眼罩 == 0
			PRINT 被自己的甜膩愛液弄髒的、
		PRINT 染上淫靡污漬的、
	;淫乱or施虐狂or施虐属性Lv.5以上で「害羞」を持たない
	;ELSEIF (TALENT:淫乱 || TALENT:小悪魔 || ABL:施虐属性 >= 5) && TALENT:害羞 == 0
	;	PRINT 滴落著透明唾液的
	ELSEIF RAND:3 == 0
		PRINTFORM 散發著異臭的
	ELSE
		PRINTFORM %TEXTS("阴茎：視覚", PLAYER)%
	ENDIF
ELSE
	;六九式
	IF TEQUIP:六九式
		PRINTFORM %TEXTS("６９")%
	;眼罩で分岐。ところで、Ｍ字開脚座りってもっといい名称無いですかね？
	ELSEIF TEQUIP:眼罩 || TEQUIP:拘束
		PRINT 、被嘴唇慢慢摸索到位置的
	ENDIF

	;直前の調教と汚れで分岐
	IF PREVEX("放尿", PLAYER)
		IF RAND:2
			PRINTFORM 散發著撲鼻而來的強烈臭味的
		ELSE
			PRINTFORM 殘留著濁黃尿液的
		ENDIF
	ELSEIF PREVPLAY("挿入中", PLAYER)
		CALL ADD_TEXT, "性器から口へ"
	ELSEIF RAND:3 == 0
		PRINTFORM 散發著異臭的
	ELSE
		PRINTFORM %TEXTS("阴茎：視覚", PLAYER)%
	ENDIF
ENDIF

PRINTFORM 陰莖直到根部都%TEXTR("一口氣/深深吸入/輕而易舉地")%吞入咽喉、
IF TALENT:恋慕 && RAND:4 == 0
	IF RAND:2 && COND("陰毛", PLAYER) >= 5 && TEQUIP:六九式 == 0
		PRINTFORM 將鼻尖埋入陰毛中細嗅摩擦著的同時、
	ELSE
		PRINTFORM 因口中的觸感與味道而陶醉的同時
	ENDIF
ELSEIF ABL:奉仕精神 >= 3 && TEQUIP:推倒 == 0 && TEQUIP:六九式 == 0 && RAND:3 == 0
	PRINTFORM 用%TEXTR("不住上翻的眼珠/濕潤的瞳孔")%仰視著神情的同時、
ELSEIF ABL:技巧 >= 3 && TEQUIP:口枷 == 0 && RAND:2
	PRINTFORM %TEXTR("控制著緩急/用雙唇緊緊捆緊")%的同時、
ELSEIF TALENT:嘔吐反射耐性 == 0 && RAND:2
	PRINTFORM 無法抑制的肉體反射讓瞳孔痛苦的濕潤的同時、
ELSE
	PRINTFORM 從嘴角%TEXTR("充溢這細末的白沫的唾液/黏稠發亮的大量涎水")%慢慢垂落的同時、
ENDIF
;技巧４かつ奉仕精神３
IF ABL:技巧 >= 4 && ABL:奉仕精神 >= 3
	IF RAND:2
		PRINTFORMW 將整個咽喉驅使著刺激深入口中的陰莖……
	ELSE
		PRINTFORMW 用輕輕抽搐的喉嚨擠壓並吮吸著陰莖……
	ENDIF
ELSEIF RAND:2
	PRINTFORMW 用不住發出嗚咽的喉嚨盡力吮吸著……
ELSE
	PRINTFORMW 用喉咙\@ TEQUIP:推倒 ? 責弄 # 奉仕 \@着…
ENDIF

;TEQUIP:深喉の値で誰のものを咥えているかを記録
TEQUIP:深喉 = PLAYER
;フラグとTEQUIPが分かれているのは、射精後などにTEQUIP:深喉が０となっている時でも参照するため
CALL SETFLAG, "深喉フラグ", TARGET

@COM_ORDER_COM150, ARG
#DIM MEMO_STAIN
VARSET LOCAL
LOCALS =

;すべての命令に共通の要素を考慮
;(信頼が高いと命令に従いやすいなど)
CALL COM_ORDER, ARG

LOCAL = RESULT

;ABL:奉仕精神
IF ABL:奉仕精神
	LOCAL += MIN(ABL:奉仕精神, 10) * 2
	LOCALS = %LOCALS% + 奉仕精神LV{ABL:奉仕精神}({MIN(ABL:奉仕精神, 10) * 2})
ENDIF
;ABL:受虐属性
IF ABL:受虐属性
	LOCAL += MIN(ABL:受虐属性, 5) * 3
	LOCALS = %LOCALS% + 受虐属性LV{ABL:受虐属性}({MIN(ABL:受虐属性, 5) * 3})
ENDIF
;ABL:精液中毒
IF ABL:精液中毒
	LOCAL += MIN(ABL:精液中毒, 5) * 3
	LOCALS = %LOCALS% + 精液中毒LV{ABL:精液中毒}({MIN(ABL:精液中毒, 5) * 3})
ENDIF
;害羞
IF TALENT:害羞
	LOCAL -= 1
	LOCALS = %LOCALS% - 害羞(1)
ENDIF
;献身的
IF TALENT:献身
	LOCAL += 6
	LOCALS = %LOCALS% + 献身(6)
ENDIF
;否定快感
IF TALENT:否定快感
	LOCAL -= 1
	LOCALS = %LOCALS% - 否定快感(1)
ENDIF
;Ｍ性向
IF TALENT:Ｍ性向
	LOCAL += 10
	LOCALS = %LOCALS% + Ｍ性向(10)
ENDIF
;嘔吐反射耐性
IF TALENT:嘔吐反射耐性
	LOCAL += 20
	LOCALS = %LOCALS% + 嘔吐反射耐性(20)
ENDIF

MEMO_STAIN = COM_ORDER_STAIN("Ｐ", "Ｍ")
;施虐狂なら、汚れていても、かまわずに突っ込みます
SIF ABL:PLAYER:施虐属性 >= 3
	MEMO_STAIN /= 3

;汚れ有
IF MEMO_STAIN
	LOCAL -= MEMO_STAIN
	LOCALS = %LOCALS% - 汚れ有
	;汚臭鈍感
	SIF TALENT:汚臭鈍感 || TALENT:汚臭敏感
		LOCALS = %LOCALS%、\@TALENT:汚臭鈍感 ? 汚臭鈍感 # 汚臭敏感\@
	SIF COND("陶酔")
		LOCALS = %LOCALS%、陶酔で半減
	LOCALS = %LOCALS%({MEMO_STAIN})
ENDIF

;合計を表示(60以上で実行。強制口交よりも高難易度)
CALL PRINT_COM_ORDER_COM, LOCALS, ARG, LOCAL, 60

RETURN RESULT


code:
@COM151
;尿道探针
;何度も実行したくないので、消耗は激しいが効果は高い指令にしたい
CALL PRINT_TRAIN_NAME(SELECTCOM)

;-------------------------------------------------
;実行できるかの判定
;-------------------------------------------------
CALL SHOW_COM_ORDER_COM, 151

;実行できない
SIF TCVAR:干劲 == 0
	RETURN 0
;-------------------------------------------------
;実行決定
;-------------------------------------------------
CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

RETURN 1

@FLAG_COM151, ARG
CALL SETFLAG, "尿道性感フラグ", TARGET
;尿的污垢発生
IF PENIS(TARGET)
	STAIN:Ｐ = SET_STAIN(STAIN:Ｐ, "尿")
ELSE
	STAIN:Ｖ = SET_STAIN(STAIN:Ｖ, "尿")
ENDIF
;胖次をずらす
SIF !CHECK_CLO("Ｖ露出") && INSERT("挿入") == 0
	CALL SETFLAG, "ボトムを露出：テ接吻ト没有", TARGET

@SOURCE_COM151, ARG
;くぱぁ
CALL SOURCE_COM13, 50
CALL SOURCE_INSERTION, "尿道探针", ARG, 100

;尿道探针の太さを選択する
@SELECT_BOUGIE
#DIM LCOUNT
#DIM SELECT_BOUGIE
#DIM BOUGIE, 10
#DIM CHOICE

;どっちの尿道か
LOCALS = %CONDS("尿道拡張度")%

;今までに経験した太さが、疼痛のある太さかどうか。最小*1.05…{MIN_STR(LOCALS)*21/20}mmまでなら疼痛は少ない
;例えば、LOCALS = 10/10/10/9/9/9/9/8/8/8 ならMAX_STR(LOCALS) = 10, CALCF("尿道径") = 9, MIN_STR(LOCALS) = 8 となる
;ちなみにMAX_STR(LOCALS) = MIN_STR(LOCALS) + 1
;初回
IF COND("尿道履歴") == 0
	PRINTFORML %CALLNAME:TARGET%的尿道、完全沒有接受過任何異物的侵入。
	PRINTFORML 今回{MIN_STR(LOCALS)}mm、甚至更粗的{MIN_STR(LOCALS) + 1}mm尝试一下似乎也是可以的
	PRINTFORML ……要使用什麼粗細的尿道探針？
;ブジーよりも粗大的ものを入れられたことが有そう
ELSEIF MAX_STR(LOCALS) > COND("尿道履歴")
	PRINTFORML %CALLNAME:TARGET%曾經使用過的尿道探針中最粗的記錄應該是{COND("尿道履歴")}mm……但是、要想使用比那一次更粗的東西的話似乎也不是不能。
	PRINTFORML ……要使用什麼粗細的尿道探針？
ELSE
	SELECTCASE MAX_STR(LOCALS) - MIN_STR(LOCALS)*21/20
	CASE IS >= 1
		PRINTFORML %CALLNAME:TARGET%的尿道对于直径{MAX(MAX_STR(LOCALS), COND("尿道履歴"))}mm的已经経験済了、对于{MIN_STR(LOCALS)*21/20}mm的疼痛应该会较少吧
		PRINTFORML {MAX(MIN_STR(LOCALS)*21/20, MIN_STR(LOCALS) + 1)}mm以下的粗细还是比较好的……\@ PREVCOM == SELECTCOM ? 接下来 # 这次 \@、要使用什麼粗細的尿道探針？
	CASE 0
		PRINTFORML %CALLNAME:TARGET%的尿道对于直径{MAX(MAX_STR(LOCALS), COND("尿道履歴"))}mm的已经経験済了、按照目前来看似乎不会有太大疼痛的样子
		PRINTFORML {MAX(MAX_STR(LOCALS), COND("尿道履歴"))}mm以下的粗细还是比较好的……\@ PREVCOM == SELECTCOM ? 接下来 # 这次 \@、要使用什麼粗細的尿道探針？
	CASEELSE
		PRINTFORML %CALLNAME:TARGET%的尿道对于直径{MAX(MAX_STR(LOCALS), COND("尿道履歴"))}mm的已经経験済了、这个粗细是没问题的
		PRINTFORML 看来挑戦{MAX_STR(LOCALS) + 1}mm也不会有什么問題……\@ PREVCOM == SELECTCOM ? 接下来 # 这次 \@、要使用什麼粗細的尿道探針？
	ENDSELECT
ENDIF

;連続実行だと２倍行う
IF PREVCOM == SELECTCOM
	SETCOLOR DEF_COLOR("黄色")
	PRINTFORML ※因為是對同一指令的連續實行、將對%CALLNAME:TARGET%的尿道進行更加仔細地蹂躪※
	RESETCOLOR
;肉体的な限界がある
ELSEIF CALCF("尿道径") >= CALCF("尿道拡張の壁")
	SETCOLOR DEF_COLOR("黄色")
	IF CALCF("尿道径") >= CALCF("尿道拡張の限界")
		PRINTFORML ※%CALLNAME:TARGET%的尿道已经到极限了……在這之上的擴張已經是不可能的強人所難了※
	ELSE
		PRINTFORML ※%CALLNAME:TARGET%的尿道已經非常接近極限了……在這之上的擴張應該是非常困難的了※
	ENDIF

	SETCOLOR DEF_COLOR("爱心粉紅")
	;ちょっとお得な情報
	CALL COM_ABLE146
	IF RESULT && COND("身嗜み：尿道拡張") == 0 && ABL:尿道拡張 >= 0 && EXP:尿道経験 >= 5 && (EXP:尿道経験 >= 20 || EXP:尿道拡張経験 >= 5 || ABL:尿道拡張 >= 1)
		PRINTFORML 介於之前的身嗜管理中下達的指示、尿道的擴張已經能由%CALLNAME:TARGET%自己進行了。
		PRINTFORML 儘管很難與%CALLNAME:PLAYER%毫不留情的調教手段相比、如何呢？
	ENDIF
	RESETCOLOR
ENDIF
PRINTL

;MIN_STR(LOCALS)mmの太さは拡張という点では弱いが、尿道経験は増えるので一応意味はある
SELECT_BOUGIE = MAX(MAX_STR(LOCALS) - MIN_STR(LOCALS), 3) + MIN_STR(LOCALS)/20
SIF ABL:PLAYER:施虐属性 >= 5 || ABL:受虐属性 >= 5
	SELECT_BOUGIE += 1
SIF ABL:PLAYER:施虐属性 >= 3
	SELECT_BOUGIE += 1
;調教知識によるBOUNS
SIF COND("性器拡張知識", PLAYER)
	SELECT_BOUGIE += 1
;太さの1.2倍までとする…最大値があまりに低い場合は３保障
SELECT_BOUGIE = MAX(LIMIT(SELECT_BOUGIE, 3, MIN_STR(LOCALS)/5 + 1), 3, MIN_STR(LOCALS)/10 + 1)

VARSET LOCAL
VARSET BOUGIE

FOR LCOUNT, 0, SELECT_BOUGIE
	;選択肢が多すぎる場合には削る
	SELECTCASE MIN_STR(LOCALS) + LCOUNT
	CASE IS >= 40
		SIF SELECT_BOUGIE >= 10 && (MIN_STR(LOCALS) + LCOUNT)%5
			CONTINUE
	ENDSELECT

	BOUGIE:LOCAL = MIN_STR(LOCALS) + LCOUNT
	PRINTFORML  [{LOCAL, 2}] {BOUGIE:LOCAL, 2}mm

	LOCAL += 1
NEXT

PRINTFORML  [99] 查看補充說明
PRINTFORML [100] 不、再考慮一下吧。

CALL INPUT_SELECT, LOCAL, 99, 100

CHOICE = RESULT

SELECTCASE CHOICE
CASE 100
	TFLAG:指令中止 = 1
	RETURN 0
CASE 99
	PRINTFORML □ 補足 □---------------------------------------------------------------------
	PRINTFORM 在對尿道進行擴張的時候、如果具有
	CALL PRINT_COLORTEXT, "[性器拡張知識]", DEF_COLOR("黄色")
	PRINTFORM 或
	CALL PRINT_COLORTEXT, "[拡張適性]", DEF_COLOR("黄色")
	PRINTFORML 的话会更容易出現進展。
	PRINTFORM 尽管并不是对拡張的容易度有影響的素質、
	CALL PRINT_COLORTEXT, "[倒錯的]", DEF_COLOR("黄色")
	PRINTFORML 有这些的话就可以抑制反感
	PRINTFORML

	PRINTFORML 尿道產生的快感與尿道感覺是相互依存的……比如、在初期疼痛感很強的時候幾乎不可能獲得快感。
	PRINTFORML 随着経験的累积、痛苦会減少、习惯被玩弄尿道之后就没问题了
	PRINTFORM 积累了因为玩弄尿道而絶頂的経験之后、
	CALL PRINT_COLORTEXT, "[尿道性感]", DEF_COLOR("黄色")
	PRINTFORML 就可以觉醒出来了
	CALL PRINT_COLORTEXT, "[尿道性感]", DEF_COLOR("黄色")
	PRINTFORMW 没有的话就基本上不怎么能变舒服、所以请尽早取得
	PRINTFORML -------------------------------------------------------------------------------
CASEELSE
	IF BOUGIE:CHOICE
		TEQUIP:尿道探针 = BOUGIE:CHOICE
		;今までに使われた最大の尺寸を記録
		IF TEQUIP:尿道探针 > COND("尿道履歴")
			IF PENIS(TARGET)
				CFLAG:Ｐ尿道履歴 = TEQUIP:尿道探针
			ELSE
				CFLAG:Ｖ尿道履歴 = TEQUIP:尿道探针
			ENDIF
		ENDIF
		RETURN 1
	ENDIF
ENDSELECT

PRINTL
RESTART

@TRAIN_MESSAGE_COM151
;ブジーを見た際の反応です
CALL KOJO_EVENT, "尿道探针追加口上"

;ブジーの太さ選択
CALL SELECT_BOUGIE

SIF REFUSE_CHECK()
	RETURN 0

;-------------------------------------------------
;ソースの計算(確定値)
;-------------------------------------------------
CALL CALC, "調教ソースの計算"

;地の文を飛ばすフラグを立てている(ターン開始時にOFFになる)
SIF TFLAG:地の文スキップ
	RETURN 0

;どっちの尿道か
LOCALS = %CONDS("尿道拡張度")%

PRINTFORM %CALLNAME:PLAYER%將%CALLNAME:TARGET%的
IF GETBIT(TEQUIP:筋弛緩剤, 1)
	PRINTFORM 因為藥物而鬆弛下來的尿道
ELSE
	PRINTFORM \@ PENIS(TARGET) ? 陰莖鈴口 # 尿道 \@用
ENDIF
SELECTCASE TEQUIP:尿道探针
CASE IS < 15
	PRINTFORMW 用塗上了果凍的{TEQUIP:尿道探针}mm探針挿入了。
;指並み
CASE 15 TO 25
	PRINTFORMW 用手指粗细的{TEQUIP:尿道探针}mm探針挿入了。
;短小阴茎並み
CASE 26 TO 32
	PRINTFORMW 用已经可以说是较細的阳具直径的{TEQUIP:尿道探针}mm探針挿入了。
;ふつうのちんこ並み
CASE 33 TO 39
	PRINTFORMW 用一般情况下根本不可能进入的{TEQUIP:尿道探针}mm探針挿入了。
;巨根並み～
CASEELSE
	PRINTFORMW 直径{TEQUIP:尿道探针}mm的光滑金属製阳具挿入了。
ENDSELECT

;EXP:尿道経験、TALENT:尿道性感、TALENT:倒錯的、TEQUIP:尿道探针 > CALCF("尿道径") あたりで分岐か
;拡張する段階ごとに中間報告（太さを飛ばしてると出ないが些細な事だろう）
IF TEQUIP:尿道探针 > MAX_STR(LOCALS) && CALCF("尿道拡張の限界") >= TEQUIP:尿道探针 + 2
	SELECTCASE MIN_STR(LOCALS)
	CASE 13
		IF CHOSECOM(SELECTCOM) == 0
			PRINTFORML ……%CALLNAME:TARGET%的尿道、被粗大的玩具擴張了。
			PRINTFORMW 在尿道回縮之前、手指粗細的東西也許也能接受了吧……
			RETURN 1
		ENDIF
	CASE 18
		IF CHOSECOM(SELECTCOM) == 0
			PRINTFORML ……%CALLNAME:TARGET%的尿道、似乎已經被擴張的相當鬆弛了。
			PRINTFORMW 或許、在努力一下就連普通粗細的男性器也能接受了吧……
			RETURN 1
		ENDIF
	CASE 25
		IF CHOSECOM(SELECTCOM) == 0
			PRINTFORML ……%CALLNAME:TARGET%的尿道、似乎已經被擴張的相當鬆弛了。
			PRINTFORMW 或許、在努力一下就連普通粗細的男性器也能接受了吧……
			RETURN 1
		ENDIF
	CASE 32
		IF CHOSECOM(SELECTCOM) == 0
			PRINTFORML ……%CALLNAME:TARGET%的尿道、已經完全化作另一個大張著的肉穴。
			PRINTFORMW 現在、恐怕就連性行為也能沒問題的直接進行吧……
			RETURN 1
		ENDIF
	ENDSELECT
ENDIF

;初めてのブジー受け入れ
IF FIRSTCOM(SELECTCOM) == 0 && EXP:尿道拡張経験 == 0
	;初回から拡張
	IF TEQUIP:尿道探针 > MIN_STR(LOCALS)
		PRINTFORMW 与洞口相比棒子\@ TEQUIP:尿道探针 - MIN_STR(LOCALS) >= 2 ? 過于 # 有点 \@粗了、光滑的金属棒不理会尿道括约肌的抵抗、挖掘着排尿孔强行让其适应这个粗度…
	ELSEIF PENIS(TARGET)
		PRINTFORMW 儘管對於第一次被插入錯誤的地方、但是%CALLNAME:TARGET%尿道還是將侵入的金屬小棒一直接受到抵達陰莖的根部為止…
	ELSE
		PRINTFORMW 儘管對於第一次被插入錯誤的地方、但是%CALLNAME:TARGET%尿道還是將侵入的金屬小棒一直接受到抵達膀胱為止…
	ENDIF
;各調教ごとの二回目以降のセリフ（簡潔）
ELSEIF CHOSECOM(SELECTCOM)
	IF CHECK("尿道PLAY")
		IF TALENT:尿道性感
			PRINTFORM %CALLNAME:TARGET%
			IF RAND:2
				PRINTFORM 將深深插入體內的金屬棒用\@ PENIS(TARGET) ? 陰莖的根部 # 膀胱 \@輕鬆地接受了、
			ELSE
				PRINTFORM 那原本只有液體通過的小孔\@ TALENT:尿道性感 ? 歡喜地 # \@將異物吞入體內、
			ENDIF
			IF TALENT:尿道性感 && RAND:3 == 0
				PRINTFORMW 在臉頰因快樂與羞恥而佈滿紅潮的同時、盡情享受著尿液隨著金屬棒的插拔飛濺而出的下流PLAY……
			ELSEIF RAND:2
				PRINTFORMW 在金屬棒抽插的同時配合的將括約肌繃緊又或放鬆……
			ELSE
				PRINTFORMW 在敏感粘膜被金屬小棒刮擦的時候、表現出令人興奮起來的淫亂反應……
			ENDIF
		ELSEIF PENIS(TARGET)
			PRINTFORMW 完全沒有受到任何阻礙的強硬闖入最深處的金屬小棒、在激烈地進出之時也將陰莖的海綿體不斷推開、讓%CALLNAME:TARGET%將尿道被異物侵入的感覺與由此而生的奇妙快感深深銘記下來……
		ELSE
			PRINTFORMW 就像滑动一様毫无阻碍地直达膀胱的金属小棒、在用進出時漏出的體液將地面的積水擴大的同時、讓%CALLNAME:TARGET%將尿道被異物侵入的感覺與由此而生的奇妙快感深深銘記下來……
		ENDIF
	ELSE
		PRINTFORMW 強烈地抵抗著的小小孔洞、在金屬材質的擴張探針的作用下、緩慢而切實地向另一個性器推進著……
	ENDIF
;拡張
ELSEIF TEQUIP:尿道探针 > MIN_STR(LOCALS)
	;疼痛や抵抗が少ない
	IF CHECK("尿道PLAY")
		IF TALENT:尿道性感 && RAND:2
			PRINTFORMW 已經完全作為性器覺醒了的%CALLNAME:TARGET%的尿道、不但將侵入的異物毫不在意地\@ PENIS(TARGET) ? 將20cm的長度完全吞下 # 用膀胱輕易接受了 \@、而且還在抽出時戀戀不捨的緊緊吮吸著……
		ELSE
			PRINTFORMW 完全沒有任何反抗之意的接受了金屬棒的小孔、順從著毫無空隙的擴張活動而被反復拉扯著柔軟的肉襞……
		ENDIF
	ELSEIF EXP:尿道経験 >= 20 || EXP:尿道拡張経験 >= 20
		PRINTFORMW 早已在反復地開發下將被異物侵入的感覺習慣下來的%CALLNAME:TARGET%的尿道、在過分粗大的金屬棒所帶來的痛苦之下還是產生了強烈的反抗……
	ELSE
		PRINTFORMW 開發經驗尚且不足的、仍然只是排泄器官的尿道、在插入內部的異物所帶來的痛苦之下無法抑制地繃緊了括約肌……
	ENDIF
;拡張はしないで慣らすのみ
ELSE
	IF CHECK("尿道PLAY")
		IF TALENT:尿道性感
			PRINTFORMW %CALLNAME:TARGET%在尿道被探針肆意進出的同時、因下體傳來的快感而顫抖著腰肢、將這異常的快樂與遊戲盡情地享受著……
		ELSE
			PRINTFORMW 儘管%CALLNAME:TARGET%並沒有感到特別的疼痛、但依然挺直脊背、因難以言喻的奇妙觸感而浮現出不安的神色……
		ENDIF
	ELSEIF PENIS(TARGET)
		PRINTFORMW 在粗大而光滑的金屬棒將%CALLNAME:TARGET%的抵抗毫不在意地闖入尿道的最深處、在激烈地插拔著將被敏感粘膜包裹著的狹窄通道擴張開來的同時、也將這異常的感觸深深地銘刻下來……
	ELSE
		PRINTFORMW 在粗大而光滑的金屬棒將%CALLNAME:TARGET%的抵抗毫不在意地闖入膀胱、激烈地插拔著、將被敏感粘膜包裹著的狹窄通道隨著尿液的飛濺而出擴張開來……
	ENDIF
ENDIF

@COM_ORDER_COM151, ARG
VARSET LOCAL
LOCALS =

;すべての命令に共通の要素を考慮
;(信頼が高いと命令に従いやすいなど)
CALL COM_ORDER, ARG

LOCAL = RESULT

;ABL:露出癖
IF ABL:露出癖
	LOCAL += MIN(ABL:露出癖, 10) * 2
	LOCALS = %LOCALS% + 露出癖LV{ABL:露出癖}({MIN(ABL:露出癖, 10) * 2})
ENDIF
;ABL:受虐属性
IF ABL:受虐属性
	LOCAL += MIN(ABL:受虐属性, 5) * 3
	LOCALS = %LOCALS% + 受虐属性LV{ABL:受虐属性}({MIN(ABL:受虐属性, 5) * 3})
ENDIF
;冷静
IF TALENT:冷静
	LOCAL -= 5
	LOCALS = %LOCALS% - 冷静(5)
ENDIF
;害羞
IF TALENT:害羞
	LOCAL -= 15
	LOCALS = %LOCALS% - 害羞(15)
ENDIF
;不知羞恥
IF TALENT:不知羞恥
	LOCAL += 2
	LOCALS = %LOCALS% + 不知羞恥(2)
ENDIF
;清楚
IF TALENT:清楚
	LOCAL -= 10
	LOCALS = %LOCALS% - 清楚(10)
ENDIF
;接受快感
IF TALENT:接受快感
	LOCAL += 5
	LOCALS = %LOCALS% + 接受快感(5)
ENDIF
;否定快感
IF TALENT:否定快感
	LOCAL -= 5
	LOCALS = %LOCALS% - 否定快感(5)
ENDIF
;倒錯的
IF TALENT:倒錯的
	LOCAL += 10
	LOCALS = %LOCALS% + 倒錯的(10)
ENDIF
IF TEQUIP:拘束
	LOCAL += 30
	LOCALS = %LOCALS% + 拘束中(30)
ENDIF
;媚薬
IF TEQUIP:媚薬
	LOCAL += 2
	LOCALS = %LOCALS% + 媚薬(2)
ENDIF
;尿道性感
IF TALENT:尿道性感
	LOCAL += 10
	LOCALS = %LOCALS% + 尿道性感(10)
ELSE
	LOCAL -= 30
	LOCALS = %LOCALS% - 尿道性感没有(30)
ENDIF
;视频拍摄
IF TEQUIP:攝像機
	LOCAL -= 20
	LOCALS = %LOCALS% - 撮影中(20)
ENDIF
;野外or室内露出
IF TEQUIP:野外PLAY
	LOCAL -= 10
	LOCALS = %LOCALS% - 野外PLAY(10)
ELSEIF TEQUIP:室内露出PLAY
	LOCAL -= 10
	LOCALS = %LOCALS% - 室内露出(10)
ENDIF

;合計を表示(80以上で実行)
CALL PRINT_COM_ORDER_COM, LOCALS, ARG, LOCAL, 80

RETURN RESULT


code:
@COM152
;尿道振動棒
;0bit細い, 1bit普通, 2bit粗大的, 3bit極太, 4bit腕, 5bit馬, 11bit布ル領帯プ
CALL PRINT_TRAIN_NAME(SELECTCOM)

;-------------------------------------------------
;実行できるかの判定
;-------------------------------------------------
IF TEQUIP:尿道振動棒 == 0
	CALL SHOW_COM_ORDER_COM, 152
	;実行できない
	SIF TCVAR:干劲 == 0
		RETURN 0
ENDIF

;-------------------------------------------------
;実行決定
;-------------------------------------------------
CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

RETURN 1

;ARGに入れられる角色、ARGSは補足。
;現状だとARGS == "強制挿入"で[100] やめる を選択不可にできる
@SELECT_VIB_U, ARG, ARGS
#DIM LCOUNT
#DIM SIZE_DILDO

VARSET LOCAL, 1
;どの振動棒を使えるかを判定（双方の施虐狂受虐狂具合で変化）
FOR LCOUNT, 1, 6
	;振動棒のサイズを代入
	SIZE_DILDO = LCOUNT

	SELECTCASE SIZE_DILDO - ABL:尿道拡張
	CASE IS >= 2
		LOCAL:LCOUNT = 0
	CASE 1
		;尺寸が１上回っているなら、施虐属性３ＬＶor（施虐属性１ＬＶに加えて相手の受虐属性１ＬＶ）が必要
		SIF MAX(ABL:受虐属性, ABL:PLAYER:施虐属性) >= 3 || MIN(ABL:受虐属性, ABL:PLAYER:施虐属性) >= 1
			CONTINUE
	CASEELSE
		CONTINUE
	ENDSELECT

	LOCAL:LCOUNT = 0
NEXT

;道具所持による判定
SIF ITEM:極太振動棒 == 0
	LOCAL:3 = 0
SIF ITEM:肛門拉珠 == 0
	LOCAL:11 = 0

PRINTFORML 這一次要\@ ARG == PLAYER ? 讓她 # \@使用哪種大小的振動棒呢……？

VARSET LOCALS
LOCALS:0 = [ 0] 尿道用小型振動棒
LOCALS:1 = [\@ LOCAL:1 ? {1, 2} # × \@] 普通振動棒　　\@ ABL:ARG:尿道拡張 <= 0 ? ※初心者厳禁！ # \@
LOCALS:2 = [\@ LOCAL:2 ? {2, 2} # × \@] 大号振動棒　　\@ ABL:ARG:尿道拡張 <= 1 ? ※習慣者用！ # \@
LOCALS:3 = [\@ LOCAL:3 ? {3, 2} # × \@] 极粗振動棒　　\@ ABL:ARG:尿道拡張 <= 2 ? ※上級者専用！ # \@
LOCALS:4 = [\@ LOCAL:4 ? {4, 2} # × \@] 拳型振動棒　　\@ ABL:ARG:尿道拡張 <= 3 ? ※超上級者専用！ # \@
LOCALS:5 = [\@ LOCAL:5 ? {5, 2} # × \@] 馬型振動棒　　\@ ABL:ARG:尿道拡張 <= 4 ? ※面向不願回頭的人 # \@

LOCALS:11 = [\@ LOCAL:11 ? 11 # × \@] 尿道拉珠

FOR LCOUNT, 0, 20
	SIF STRLENS(LOCALS:LCOUNT) == 0
		CONTINUE

	SIF LOCAL:LCOUNT == 0
		SETCOLOR DEF_COLOR("灰色")
	PRINTFORML  %LOCALS:LCOUNT%
	RESETCOLOR
NEXT

SIF ARGS != "強制挿入"
	PRINTFORML [100] 不、再考慮一下吧。

$INPUT_LOOP
INPUT

IF RESULT == 100 && ARGS != "強制挿入"
	TEQUIP:ARG:尿道振動棒 = 0
	TFLAG:指令中止 = 1
	RETURN 0
ELSEIF RESULT == 0
	IF ABL:尿道拡張 <= -1
		TEQUIP:ARG:尿道振動棒 = 1p0
	ELSE
		TEQUIP:ARG:尿道振動棒 = 1p1
	ENDIF
ELSEIF RESULT == 1 && LOCAL:1
	TEQUIP:ARG:尿道振動棒 = 1p2
ELSEIF RESULT == 2 && LOCAL:2
	TEQUIP:ARG:尿道振動棒 = 1p3
ELSEIF RESULT == 3 && LOCAL:3
	TEQUIP:ARG:尿道振動棒 = 1p4
ELSEIF RESULT == 4 && LOCAL:4
	TEQUIP:ARG:尿道振動棒 = 1p5
ELSEIF RESULT == 5 && LOCAL:5
	TEQUIP:ARG:尿道振動棒 = 1p6
ELSEIF RESULT == 11 && LOCAL:11
	TEQUIP:ARG:尿道振動棒 = 1p1 + 1p11
ELSE
	GOTO INPUT_LOOP
ENDIF
RETURN 1

@TRAIN_MESSAGE_COM152
;尿道振動棒の着脱
IF TEQUIP:尿道振動棒
	PRINTFORM %CALLNAME:TARGET%\@ TEQUIP:推倒 ? は # の \@%TEXTR("尿道口/狭孔/排泄腔")%から
	IF COND("尿道布ル")
		PRINTFORMW 尿道拉珠氣勢十足的一口氣拔出了。
	ELSE
		SELECTCASE SIZE("尿道振動棒")
		CASE 5
			PRINTFORM 幾乎不可能進入的馬型振動棒
		CASE 4
			PRINTFORM 成年男性手腕粗細的拳型振動棒
		CASE 3
			PRINTFORM 常規而言最為粗壯的振動棒
		CASE 2
			PRINTFORM 粗大的振動棒
		CASE 1
			PRINTFORM 尿道振動棒
		CASE 0
			PRINTFORM 細小的振動棒
		CASE -1
			PRINTFORM 和尾指相比也略顯纖細的珍袖振動棒
		ENDSELECT
		SIF ABL:MASTER:施虐属性 >= 3 && TEQUIP:推倒 == 0
			PRINTFORM 快速地
		PRINTFORMW 拔出了。

		SELECTCASE ABL:尿道拡張
		CASE IS >= 2
			PRINTFORMW ……%CALLNAME:TARGET%的尿道口欲求不滿的張合著、從沒能徹底閉上的腔口中、渾濁的濃郁滴滴答答的流下了。
			BASE:尿意 = MAX(BASE:尿意 - 2000, 0)
		CASEELSE
			PRINTFORMW ……%CALLNAME:TARGET%的尿道在擴張之的影響向稍稍地鬆弛了一些、在振動棒抽出之時也帶出了一點點的尿液。
			BASE:尿意 = MAX(BASE:尿意 - 500, 0)
		ENDSELECT
	ENDIF

	SIF COND("尿道布ル")
		CALL SETFLAG, "尿道布ル引き抜きフラグ", TARGET

	TFLAG:地の文分岐 = TEQUIP:尿道振動棒
	TEQUIP:尿道振動棒 = 0
	CALL CALC, "調教ソースの計算"
	RETURN 0
ENDIF

$SELECT_VIB
;振動棒選択
CALL SELECT_VIB_U, TARGET

SIF REFUSE_CHECK()
	RETURN 0

IF TEQUIP:尿道振動棒 == 0
	TFLAG:地の文分岐 = 1
	GOTO SELECT_VIB
ENDIF

;-------------------------------------------------
;ソースの計算(確定値)
;-------------------------------------------------
CALL CALC, "調教ソースの計算"

;地の文を飛ばすフラグを立てている(ターン開始時にOFFになる)
SIF TFLAG:地の文スキップ
	RETURN 0

PRINTFORM %CALLNAME:PLAYER%將%CALLNAME:TARGET%的
;尿道が露出していないで、胖次をずらしていないなら自動的にずらす
IF !CHECK_CLO("Ｖ露出")
	CALL SETFLAG, "ボトムを露出：テ接吻ト有些", TARGET
	PRINT 、
ENDIF
IF PREVEX("尿道絶頂") && RAND:3 == 0
	PRINTFORM 在絕頂之後%TEXTR("劇烈顫抖著的/淫靡的顫動著的/反復地張合著的/痙攣著的")%
ELSEIF PREVEX("尿道内射精")
	PRINTFORM %TEXTR("不斷溢出精液的/緩緩流出精液的/被精液完全弄髒的/沾滿精液的")%
ELSEIF (ABL:尿道拡張 >= 4 || (PREVCOM >= 83 && PREVCOM <= 88) || PREVCOM == 15) && ABL:尿道拡張 >= 2 && RAND:3 == 0
	IF PREVCOM >= 83 && PREVCOM <= 88 && RAND:2 == 0
		PRINT 在被激烈侵犯之後難以復原的
	ELSE
		PRINT 不檢點地大張著洞口的
	ENDIF
ELSEIF ABL:尿道拡張 >= 2
	PRINT 在充分地開發下十分柔軟的
ELSEIF TALENT:恋慕 || RAND:3 == 0
	PRINT 可愛的
ELSEIF EXP:尿道経験 < 50 && RAND:2 == 0
	PRINT 經驗不足的
ELSE
	PRINT 緊張地收縮著的
ENDIF
PRINT 尿道
IF TALENT:恋慕
	PRINTFORM 溫柔地揉搓著放鬆下來之後\@ COND("ボトムを露出させた") ? # 、 \@用
ELSE
	PRINTFORM に
ENDIF

IF COND("尿道布ル")
	PRINTFORMW 一粒粒分開的尿道拉珠填滿了。
ELSE
	SELECTCASE SIZE("尿道振動棒")
	CASE 5
		PRINTFORM 幾乎不可能進入的馬型
	CASE 4
		PRINTFORM 成年男性手腕粗細的
	CASE 3
		PRINTFORM 极粗的
	CASE 2
		PRINTFORM 巨大的
	CASE 1
		PRINTFORM 尿道
	CASE 0
		PRINTFORM 細小的
	CASE IS < 0
		PRINTFORM 和尾指相比也略顯纖細的珍袖
	ENDSELECT
	PRINT 振動棒
	;拡張不足
	IF ABL:尿道拡張 < SIZE("尿道振動棒")
		PRINTFORMW %TEXTR("用力地/強硬的/毫不留情地/在不斷開闢膣道的同時/在將肉體強硬推開的感觸仔細品味著的同時/將所有的抵抗毫不在意地無視的同時")%塞滿了。
	ELSE
		IF CHECK("尿道振動棒") == 0 && ABL:PLAYER:施虐属性 >= 3 && RAND:2 == 0
			PRINT 用力地
		ELSEIF (TALENT:恋慕 || ABL:施虐属性 >= 3) && RAND:3 == 0
			PRINT 深深地
		ELSEIF TALENT:恋慕 && RAND:3 == 0
			PRINT 温柔地
		ELSEIF (TALENT:恋慕 || ABL:施虐属性 >= 3) && RAND:2 == 0
			PRINT 一口気
		ELSE
			PRINT 溫柔緩慢地
		ENDIF
		IF RAND:4 == 0
			PRINTW 填滿了內部。
		ELSEIF CHECK("尿道振動棒") == 0 && ABL:PLAYER:施虐属性 >= 3 && RAND:2 == 0
			PRINTW 衝入其中。
		ELSEIF (TALENT:恋慕 || ABL:施虐属性 >= 3) && RAND:3 == 0
			PRINTW 沉入其中。
		ELSEIF (TALENT:恋慕 || ABL:施虐属性 >= 3) && RAND:2 == 0
			PRINTW 刺入其中。
		ELSE
			PRINTW 挿入其中。
		ENDIF
	ENDIF
ENDIF

;拡張十分
IF CHECK("尿道振動棒") && CHOSECOM(SELECTCOM) == 0
	PRINTFORM %CALLNAME:TARGET%的
	IF TALENT:娇小 && EXP:尿道経験 < 100 && ABL:尿道拡張 <= 1
		PRINT 小小的
	ELSEIF TALENT:娇小 && ABL:尿道拡張 >= 2
		SIF ABL:尿道拡張 > SIZE("尿道振動棒")
			PRINT 完全
		PRINT 被充分擴張的
	ELSEIF EXP:尿道経験 >= 100 && RAND:2 == 0
		PRINT 経験豊富的
	ELSEIF ABL:尿道拡張 > SIZE("尿道振動棒")
		PRINT 十分柔軟的
	ENDIF
	PRINT 尿道
	IF SIZE("尿道振動棒") >= 2 && ABL:尿道拡張 > SIZE("尿道振動棒") && RAND:2
		IF RAND:3 == 0
			PRINT 輕而易舉地
		ELSEIF RAND:2
			PRINT 毫無抵抗地
		ELSE
			PRINT 就像是在主動吮吸一樣地
		ENDIF
	ELSEIF ABL:尿道拡張 >= SIZE("尿道振動棒") + 1 && RAND:2 == 0
		SIF TALENT:内気 || TALENT:清楚
			PRINT 與平日裡清純老實的印象完全不同、
		PRINT 輕易地張開、
	ELSEIF ABL:尿道拡張 >= SIZE("尿道振動棒")
		PRINT 向橡膠一樣伸展開來、
	ENDIF
	IF COND("尿道布ル")
		PRINT 將尿道拉珠
	ELSE
		SELECTCASE SIZE("尿道振動棒")
		CASE IS >= 3
			IF TALENT:男性 || TALENT:処女
				PRINT 將根本沒法用嘴含住的
			ELSEIF ABL:Ｖ拡張 < SIZE("尿道振動棒")
				PRINT 將就連前面的膣穴也無法接納的
			ENDIF
			PRINT 兇惡的
		CASE 2
			PRINT 粗大的
		CASE IS <= 0
			PRINT 將尿道
		ENDSELECT
		PRINT 振動棒
	ENDIF
	IF RAND:2 == 0
		PRINTFORMW 用尿道整個吞嚥下來……
	ELSE
		PRINTFORMW 一直吞到膀胱的內壁為止……
	ENDIF
ELSEIF CHECK("尿道振動棒")
	IF COND("尿道布ル")
		PRINTFORMW 在尿道乃至於膀胱都被異物填滿所帶來的奇妙感觸、讓%CALLNAME:TARGET%無法抑制的漏出了灼熱而濡濕的低吟聲……
	ELSE
		PRINTFORMW 在%TEXTR("尿道口/狭孔/排泄腔")%中激烈震動的振動棒所傳來的異常快感、讓%CALLNAME:TARGET%無法抑制的漏出了灼熱而濡濕的低吟聲……
	ENDIF
	;道端の菠萝包加筆分
	SIF ABL:尿道拡張 - SIZE("尿道振動棒") >= 2
		PRINTFORMW しかし拡張されつくした%CALLNAME:TARGET%には小さ過ぎるようで、その声にはどこか物足りなさげな色が浮かんでいる…
ELSEIF CHECK("尿道振動棒") == 0 && ABL:尿道拡張 < SIZE("尿道振動棒")
	PRINTFORM %CALLNAME:TARGET%的
	SIF TALENT:娇小 && EXP:尿道経験 < 100 && ABL:尿道拡張 == 0
		PRINT 小小的
	PRINT 尿道
	IF SIZE("尿道振動棒") >= 1 && ABL:尿道拡張 < SIZE("尿道振動棒")
		PRINTW 無法承受振動棒的粗大、無法抑制的吐出了愈來愈高的哀鳴聲……
	ELSE
		PRINT 被強硬地擴張到幾乎要裂開的程度、但終究將
		IF COND("尿道布ル")
			PRINT 相互連接著的球體
		ELSEIF SIZE("尿道振動棒") == 0
			PRINT 尿道振動棒
		ELSE
			PRINT 振動棒
		ENDIF
		PRINTW 勉強地吞入體內……
	ENDIF
ELSEIF ABL:尿道感覚 >= 2 && CUP:苦痛 < 500
	PRINTFORMW %CALLNAME:TARGET%因腸道中傳來的奇妙感觸而迷惑了……
ELSE
	PRINTFORM 對於
	IF COND("尿道布ル")
		PRINT 尿道拉珠
	ELSE
		PRINT 振動棒
	ENDIF
	PRINTFORMW 所帶來的刺激還不習慣、%CALLNAME:TARGET%的%TEXTR("尿道口/狭孔/排泄腔")%本能的緊繃著、表現著強烈的反抗心……
ENDIF

@FLAG_COM152, ARG
CALL FLAG_INSERTION, "尿道振動棒", ARG, 100

@SOURCE_COM152, ARG
IF TEQUIP:尿道振動棒
	IF COND("尿道布ル引き抜きフラグ")
		CALL SOURCE_INSERTION, "尿道振動棒", ARG, 30
	ELSE
		CALL SOURCE_INSERTION, "尿道振動棒", ARG, 100
	ENDIF
ELSE
	IF COND("尿道布ル引き抜きフラグ")
		CALL SOURCE_INSERTION, "尿道振動棒", ARG, 100
	ELSE
		CALL SOURCE_INSERTION, "尿道振動棒", ARG, 30
	ENDIF
ENDIF
CALL SETFLAG, "尿道性感フラグ", TARGET

@EQUIP_COM152
;尿道振動棒挿入中
SIF TEQUIP:尿道振動棒 == 0
	RETURN 0
SIF SELECTCOM == 152
	RETURN 0

CALL SOURCE_COM152, 30
CALL FLAG_COM152, 30

RETURN 1

@EQUIP_COM152_2
;尿道振動棒挿入中
SIF TEQUIP:尿道振動棒 == 0
	RETURN 0
SIF SELECTCOM == 152
	RETURN 0

IF COND("尿道布ル")
	PRINTL ＜尿道拉珠挿入中＞
ELSE
	SELECTCASE SIZE("尿道振動棒")
	CASE 6
		PRINTL ＜尿道規格外振動棒挿入中＞
	CASE 5
		PRINTL ＜尿道馬型振動棒挿入中＞
	CASE 4
		PRINTL ＜尿道腕並振動棒挿入中＞
	CASE 3
		PRINTL ＜尿道極太振動棒挿入中＞
	CASEELSE
		PRINTL ＜尿道振動棒挿入中＞
	ENDSELECT
ENDIF

LOCAL = 1
;自慰実行時は経験＋１
IF SELECTCOM == 3
	SIF ABL:尿道拡張 < SIZE("尿道振動棒")
		LOCAL += 1
ENDIF

EXP:尿道経験 += LOCAL

VARSET LOCAL
;尿道拡張経験
IF SELECTCOM == 3 || SELECTCOM == 15
	SIF SIZE("尿道振動棒") >= 2
		LOCAL = MAX(SIZE("尿道振動棒") - ABL:尿道拡張 + 1, SIZE("尿道振動棒")/3, 1)
ENDIF

IF LOCAL
	EXP:尿道拡張経験 += LOCAL
	;拡張経験による追加の疲労
	DOWNBASE:体力 += LOCAL*50
ENDIF

@COM_ORDER_COM152, ARG
VARSET LOCAL
LOCALS =

;すべての命令に共通の要素を考慮
;(信頼が高いと命令に従いやすいなど)
CALL COM_ORDER, ARG

LOCAL = RESULT

;ABL:露出癖
IF ABL:露出癖
	LOCAL += MIN(ABL:露出癖, 10) * 2
	LOCALS = %LOCALS% + 露出癖LV{ABL:露出癖}({MIN(ABL:露出癖, 10) * 2})
ENDIF
;ABL:受虐属性
IF ABL:受虐属性
	LOCAL += MIN(ABL:受虐属性, 5) * 3
	LOCALS = %LOCALS% + 受虐属性LV{ABL:受虐属性}({MIN(ABL:受虐属性, 5) * 3})
ENDIF
;害羞
IF TALENT:害羞
	LOCAL -= 15
	LOCALS = %LOCALS% - 害羞(15)
ENDIF
;不知羞恥
IF TALENT:不知羞恥
	LOCAL += 2
	LOCALS = %LOCALS% + 不知羞恥(2)
ENDIF
;接受快感
IF TALENT:接受快感
	LOCAL += 5
	LOCALS = %LOCALS% + 接受快感(5)
ENDIF
;否定快感
IF TALENT:否定快感
	LOCAL -= 5
	LOCALS = %LOCALS% - 否定快感(5)
ENDIF
;倒錯的
IF TALENT:倒錯的
	LOCAL += 10
	LOCALS = %LOCALS% + 倒錯的(10)
ENDIF
IF TEQUIP:拘束
	LOCAL += 30
	LOCALS = %LOCALS% + 拘束中(30)
ENDIF
;媚薬
IF TEQUIP:媚薬
	LOCAL += 2
	LOCALS = %LOCALS% + 媚薬(2)
ENDIF
;尿道性感
IF TALENT:尿道性感
	LOCAL += 10
	LOCALS = %LOCALS% + 尿道性感(10)
ENDIF
;视频拍摄
IF TEQUIP:攝像機
	LOCAL -= 20
	LOCALS = %LOCALS% - 撮影中(20)
ENDIF
;野外or室内露出
IF TEQUIP:野外PLAY
	LOCAL -= 10
	LOCALS = %LOCALS% - 野外PLAY(10)
ELSEIF TEQUIP:室内露出PLAY
	LOCAL -= 10
	LOCALS = %LOCALS% - 室内露出(10)
ENDIF

;合計を表示(55以上で実行)
CALL PRINT_COM_ORDER_COM, LOCALS, ARG, LOCAL, 55

RETURN RESULT


code:
@COM158
;写真撮影調教版
;特定の条件でのみ撮影可能
CALL PRINT_TRAIN_NAME(SELECTCOM)

CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

CALL SOURCE_CHECK_SHORT

RETURN 0

;今回のエロ中に既に撮った物と似たシチュなら保存はしない
@TRAIN_MESSAGE_COM158
#DIM LCOUNT
#DIMS EVENT_TABLE
;誰の精液か
#DIMS SPERMS

STR:写真 =
STR:補足 =

SAVESTR:写真 =
EVENT_TABLE =
SPERMS = %TEXTS("最新中出し相手の")%

PRINTFORM %CALLNAME:PLAYER%将
;射精直後は抜いて撮る　中出し後の撮影機会を増やすため
IF PREVEX("射精", PLAYER) && INSERT("挿入")
	SELECTCASE PREVCOM
	CASE TNUM("扒開阴道"), TNUM("扒開菊穴")
	CASEELSE
		PRINTFORM 把陰莖拔出、
		CALL RESET_INSERT, TARGET, PLAYER
	ENDSELECT
ENDIF
IF PREVCOM == TNUM("扒開阴道") && STR:青写真 != ""
	PRINTFORM 押開了自己的阴部的%CALLNAME:TARGET%朝向这边
	STR:写真 = %STR:青写真%
	STR:補足 += "_阴部"
ELSEIF PREVCOM == TNUM("扒開菊穴") && STR:青写真 != ""
	IF PREVPLAY("排泄")
		PRINTFORM 将出来的东西排尽直後的%CALLNAME:TARGET%朝向这边
		STR:写真 = 押開%TEXTS("お尻：感触")%、排泄直後的
		IF TALENT:縦深菊穴
			STR:写真 += "縦深菊穴被"
		ELSE
			STR:写真 += "肛門被"
		ENDIF
		STR:写真 += @"%PHOTO_SHOWING("見せつけ")%"
	ELSE
		PRINTFORM 広開了自己尻部的%CALLNAME:TARGET%朝向这边
		STR:写真 = %STR:青写真%
	ENDIF
	STR:補足 += "_肛門"
;口に出した
ELSEIF PREVEX("口内射精") && !PREVPLAY("清洁口交")
	IF PREVPLAY("自我口内射精")
		PRINTFORM 朝向刚刚在自己的口中射出精液的%CALLNAME:TARGET%
		SPERMS = 自己的
	ELSE
		PRINTFORM 对着刚刚吐出精液的%CALLNAME:TARGET%
	ENDIF
	IF ABL:精液中毒 == 0 && ABL:奉仕精神 < 5
		STR:写真 += @"因为%SPERMS%精液的味道而紧缩眉頭的"
	ELSE
		IF TEQUIP:口枷
			STR:写真 += "被道具强迫打開的口对着相机、"
		ELSE
			STR:写真 += "大张着嘴、"
		ENDIF
		IF ABL:精液中毒 || ABL:奉仕精神 >= 5 || TALENT:精液嗜好
			STR:写真 += @"证明自己将所有的%SPERMS%精液吞下去的"
		ELSE
			STR:写真 += "展示着积存在口中的%SPERMS%精液の"
		ENDIF
	ENDIF
;ハメ撮り
ELSEIF INSERT("挿入")
	PRINTFORM 保持着和%CALLNAME:TARGET%相连的体勢
	IF INSERT("正常位") || INSERT("屈曲位") || INSERT("対面座位")
		IF PREVEX("前列腺高潮射精")
			STR:写真 += @"从%TEXTR("软软的/歪歪扭扭的")%阴茎中下流地%TEXTR("漏出/流出")%精液"
		ELSEIF (INSERT("正常位") || INSERT("屈曲位")) && TEQUIP:拘束 == 0 && RAND:2
			IF (TALENT:開朗 || TALENT:喜歡引人注目 || TALENT:笨蛋) && CFLAG:夸奖 + TEQUIP:甜言蜜语
				STR:写真 += "笑着"
			ELSEIF ABL:PLAYER:施虐属性 >= 3 && RAND:3 == 0
				STR:写真 += "被单手压住"
			ELSEIF (TALENT:害羞 || TALENT:内気) && RAND:2
				STR:写真 += "両手捂住脸"
			ELSE
				STR:写真 += "在背后抓住床单"
			ENDIF
		ELSEIF ABL:欲望 >= 5 && PALAM:欲情 >= 10000 && COND("池面") == 0
			STR:写真 += @"%TEXTR("浮现出雌性的表情/嘴角流着口水")%"
		ELSE
			STR:写真 += "脸颊发红"
		ENDIF
		STR:写真 += @"ながら%CALLNAME:PLAYER%"
		IF ABL:欲望 >= 5 && RAND:3 == 0
			IF INSERT("Ａ挿入")
				STR:写真 += "进行着肛門性交在享受着の"
			ELSEIF TEQUIP:PLAYER:避孕套
				STR:写真 += "进行着以快楽为目的的性交在享受着の"
			ELSE
				STR:写真 += "进行着生性交在享受着の"
			ENDIF
		ELSEIF SIZE("阴茎", PLAYER) >= 2 && RAND:2
			IF INSERT("Ａ挿入")
				STR:写真 += @"的巨根%TEXTR("在肛門侵犯着/塞在尻穴里")%の"
			ELSE
				STR:写真 += @"的巨根%TEXTR("侵犯着/塞得慢慢的")%の"
			ENDIF
		ELSEIF INSERT("Ａ挿入")
			STR:写真 += @"%TEXTR("侵犯着肛門/在尻穴里交合着")%の"
		ELSE
			STR:写真 += @"%TEXTR("侵犯着/交合着")%の"
		ENDIF
	ELSEIF INSERT("後背位")
		SIF PREVEX("前列腺高潮射精")
			STR:写真 += @"从%TEXTR("软软的/歪歪扭扭的")%阴茎中下流地%TEXTR("漏出/流出")%精液的同时、"
		IF TEQUIP:肛用振動棒
			STR:写真 += "被振動棒挿入着的"
		ELSEIF COND("肛門栓")
			STR:写真 += "被塞子栓上的"
		ENDIF
		IF TALENT:巨尻
			STR:写真 += "大号的"
		ELSEIF TALENT:小尻
			STR:写真 += "小"
		ENDIF
		STR:写真 += @"尻%TEXTR("伸出/交出")%来供给%CALLNAME:PLAYER%侵犯の"
	ELSEIF INSERT("背面座位")
		SIF PREVEX("前列腺高潮射精")
			STR:写真 += @"从%TEXTR("软软的/歪歪扭扭的")%阴茎中下流地%TEXTR("漏出/流出")%精液的同时、"
		STR:写真 += @"坐在%CALLNAME:PLAYER%腿上被阴茎貫穿の"
	ELSEIF INSERT("騎乗位")
		IF TALENT:淫乱 + TALENT:好色 + TALENT:小悪魔 + TALENT:不知羞恥 && TEQUIP:拘束 == 0 && RANDIF(2, !TALENT:貧乳)
			IF CHECK_BODYS("兔") && RAND:3 == 0
				STR:写真 += @"双手放在頭顶上做出像是兔耳一样的手势、在%CALLNAME:MASTER%的身上%TEXTR("跳跃着/跳着下流的舞")%"
			ELSEIF EQUIP:裙 && RAND:2
				STR:写真 += "将裙子"
				IF CHECK_SKIRT("迷你") == 0 && COND("口の使用") && RAND:2
					STR:写真 += "用嘴唇叼住、"
				ELSE
					STR:写真 += "卷起来、"
				ENDIF
				IF PALAM:Ｖ潤 >= 5000 && RANDIF(1, EQUIP:胖次)
					STR:写真 += "咕啾咕啾的"
				ELSEIF EQUIP:胖次
					STR:写真 += "被挤到一边的内裤"
				ENDIF
				STR:写真 += "結合部露出来进行展示"
			ELSE
				STR:写真 += @"骑在%CALLNAME:PLAYER%上、"
				SIF PREVEX("前列腺高潮射精")
					STR:写真 += @"从%TEXTR("软软的/歪歪扭扭的")%阴茎中下流地%TEXTR("漏出/流出")%精液的同时、"
				IF ABL:露出癖 >= 3 && RAND:2
					STR:写真 += "大腿张開露出结合部の"
				ELSE
					STR:写真 += "十分享受地摇着腰の"
				ENDIF
			ENDIF
		ELSE
			STR:写真 += @"%CALLNAME:PLAYER%に突き上げられ、"
			SIF PREVEX("前列腺高潮射精")
				STR:写真 += @"%TEXTR("柔らかくなった/ふにゃふにゃの")%阴茎からだらしなく精液を%TEXTR("漏ら/垂ら")%しつつ"
			IF !TALENT:貧乳 && !TALENT:男性 && ABL:Ｂ感覚 >= 3 && TEQUIP:拘束 == 0 && RAND:2
				STR:写真 += "一边揉捏自己的胸一边喘息の"
			ELSEIF ABL:欲望 >= 3 && PALAM:欲情 >= 10000
				STR:写真 += @"带着%TEXTR("官能的/难以忍耐的/舒服的")%表情苦恼着の"
			ELSE
				STR:写真 += "脸上写满了欲望の"
			ENDIF
		ENDIF
	ENDIF
;拘束中
ELSEIF TEQUIP:拘束
	PRINTFORM 对着身體無法自由活動的%CALLNAME:TARGET%
	IF BIND("縄", TEQUIP:拘束)
		IF BIND("縄：亀甲縛り", TEQUIP:拘束)
			STR:写真 += "以龜甲縛的形式被拘束著"
		;ELSEIF BIND("縄：足吊り", TEQUIP:拘束)
		;	STR:写真 += "片足を縄で吊られ"
		ELSE
			STR:写真 += "以被繩索牢牢捆住身體的形式被拘束著"
		ENDIF
	ELSE
		IF BIND("吊り", TEQUIP:拘束)
			STR:写真 += "以手腕被與天花板連接著的鎖鏈高高吊起的形式被拘束著"
		ELSE
			IF BIND("手枷", TEQUIP:拘束)
				STR:写真 += "以雙手被手枷固定的形式"
			ELSEIF BIND("床", TEQUIP:拘束)
				STR:写真 += "以身體鎖死在床鋪上的形式"
			ELSEIF BIND("四つん這い", TEQUIP:拘束)
				STR:写真 += "以身體鎖死在方桌上的形式"
			ELSE
				STR:写真 += "以身體鎖死在專用的高台上的形式"
			ENDIF
			STR:写真 += "被拘束著"
		ENDIF
	ENDIF
	IF PREVPLAY("清洁口交")
		STR:写真 += @"たまま%CALLNAME:PLAYER%の"
		IF SIZE("阴茎", PLAYER) >= 3
			STR:写真 += "巨大阴茎"
		ELSEIF SIZE("阴茎", PLAYER) >= 2
			STR:写真 += "巨根"
		ELSE
			STR:写真 += "阴茎"
		ENDIF
		STR:写真 += "陰莖用口腔仔細清掃著の"
	ELSE
		;足開かれてる
		IF GETBIT(TEQUIP:拘束, 21)
			IF PREVPLAY("挿入")
				IF PREVEX("Ｖ中出し") || PREVEX("Ａ中出し") || PREVEX("尿道内射精")
					STR:写真 += "不斷流淌出%SPERMS%精液的"
				ELSE
					STR:写真 += "剛剛被充分侵犯過的"
				ENDIF
				IF PREVPLAY("Ｖ挿入")
					STR:写真 += "阴部"
				ELSEIF PREVPLAY("Ａ挿入")
					STR:写真 += "尻"
				ELSE
					STR:写真 += "尿道"
				ENDIF
			;多分磨鏡と指挿れの場合　女你用
			ELSE
				STR:写真 += @"%TEXTR("不斷激烈地痙攣著的/被愛液變得黏黏糊糊的")%阴部"
			ENDIF
			STR:写真 += "清楚地暴露著の"
		ELSE
			IF ABL:受虐属性 >= 3 && RAND:2
				STR:写真 += "、因興奮而使臉頰佈滿紅潮の"
			ELSEIF PREVEX("絶頂数合計") && RAND:2
				STR:写真 += "、以完全溶化的表情沉浸在絕頂的餘韻中の"
			ELSE
				STR:写真 += "、因心中的羞恥而染紅了臉頰の"
			ENDIF
		ENDIF
	ENDIF
;お掃除
ELSEIF PREVPLAY("清洁口交")
	PRINTFORM 在%CALLNAME:TARGET%在持續著侍奉的同時
	IF (TALENT:喜歡引人注目 || TALENT:開朗) && RAND:2
		STR:写真 += @"%TEXTR("对着相机比出剪刀手/用一隻手向著鏡片比出了攝像的手勢/一边对着镜頭")%、"
	ELSEIF (TALENT:好奇心 || TALENT:献身 || TALENT:小悪魔) && RAND:2
		STR:写真 += "用上翻的眼珠靜靜地凝視著鏡頭、"
	ELSEIF (TALENT:冷静 || TALENT:感情缺乏) && RAND:2
 		IF PREVEX("絶頂数合計") || COND("弛緩")
			STR:写真 += "带着放松的表情、"
		ELSE
			STR:写真 += "臉上的表情毫不動搖、"
		ENDIF
	ELSEIF (TALENT:害羞 || TALENT:内気) && RAND:2
		STR:写真 += "羞澀地將視線與照相機偏移開來、"
	ELSEIF ABL:精液中毒 <= 2 && ABL:奉仕精神 <= 2
		STR:写真 += "眉頭因不適而微微皺起、"
	ELSEIF ABL:精液中毒 >= 3 && RANDIF(1, ABL:奉仕精神 >= 3)
		STR:写真 += "因侍奉的快樂而神志恍惚、"
	ELSEIF ABL:奉仕精神 >= 3
		STR:写真 += "緊閉著雙眼一心一意的持續侍奉著、"
	ENDIF
	STR:写真 += @"%CALLNAME:PLAYER%の"
	IF PREVPLAY("恥垢清洁口交")
		STR:写真 += @"沾满恥垢的\@ SIZE("阴茎", PLAYER) >= 2 ? 巨根 # 阴茎 \@"
	ELSEIF SIZE("阴茎", PLAYER) >= 3
		STR:写真 += "巨大阴茎"
	ELSEIF SIZE("阴茎", PLAYER) >= 2
		STR:写真 += "巨根"
	ELSE
		STR:写真 += "阴茎"
	ENDIF
	STR:写真 += "在口腔内仔細清掃著の"
;避孕套で玩遊戲
ELSEIF PREVPLAY("套套内射精", PLAYER) && RAND:2
	PRINTFORM 取下避孕套
	STR:写真 += "将装滿了精液而膨脹起來的避孕套"
	IF (TALENT:小悪魔 || TALENT:好色 || TALENT:淫乱) && RAND:3 == 0
		PRINTFORM 交給%CALLNAME:TARGET%之後、
		IF COND("精液入り避孕套数") >= 2
			STR:写真 += "提在双手中展示の"
		ELSE
			IF (TALENT:喜歡引人注目|| TALENT:開朗 || ABL:露出癖 >= 3) && RAND:2
				STR:写真 += @"用嘴叼住%TEXTR("Ｗ/＿")%比出剪刀手"
			ELSE
				STR:写真 += "叼在嘴里の"
			ENDIF
		ENDIF
	ELSEIF (TALENT:好奇心 || MARK:屈服刻印 >= 2) && RAND:2
		PRINTFORM 要求%CALLNAME:TARGET%用手拿住之后、
		IF (TALENT:喜歡引人注目|| TALENT:開朗 || ABL:露出癖 >= 3) && RAND:2
			STR:写真 += "一只手捏着一只手比出剪刀手"
		ELSE
			STR:写真 += "高高舉起の"
		ENDIF
	ELSE
		PRINTFORM 放好后
		IF ABL:PLAYER:施虐属性 >= 3 && RAND:4 == 0
			STR:写真 += "在脸上"
		ELSE
			IF PREVPLAY("後背位")
				STR:写真 += "在挺直的脊背上"
			ELSE
				STR:写真 += "在緊縮的小腹上"
			ENDIF
			SIF COND("精液入り避孕套数") >= 3
				STR:写真 += "复数的"
			STR:写真 += "放置着の"
		ENDIF
	ENDIF
;０７２１
ELSEIF PREVCOM == TNUM("自慰")
	IF TALENT:淫乱 && ABL:自慰中毒 >= 5
		STR:写真 += "一边吐出舌頭露出下流的表情一边"
	ELSEIF ABL:自慰中毒 >= 3
		STR:写真 += @"%TEXTR("完全沉浸于快感中/闭着眼睛看起来很舒服地样子")%的同时"
	ELSEIF TALENT:喜歡引人注目 || TALENT:開朗 || TALENT:不知羞恥
		STR:写真 += @"一边%TEXTR("向着这边比出剪刀手/向这边展示着")%一边"
	ELSEIF TALENT:叛逆
		STR:写真 += "斜着眼看着这边的同时"
	ELSE
		STR:写真 += "害羞地闭上眼睛的同时"
	ENDIF
	IF TEQUIP:振動棒 || TEQUIP:肛用振動棒 || TEQUIP:尿道振動棒
		IF TEQUIP:振動棒
			STR:写真 += "阴部"
		ELSEIF TEQUIP:肛用振動棒
			STR:写真 += "肛門"
		ELSEIF TEQUIP:尿道振動棒
			STR:写真 += "尿道"
		ENDIF
		STR:写真 += "中的玩具用手抓住前后缓慢地抽插着"
	ELSE
		SELECTCASE FAV_ONANIE(SELECTCOM, "一番")
		CASE "Ｃ"
			STR:写真 += "用手指责弄着秘所"
		CASE "Ｖ"
			STR:写真 += "将手指插入秘裂中"
		CASE "Ａ"
		 	STR:写真 += "将手指玩弄着菊門"
		CASE "Ｂ"
			STR:写真 += "揉弄着胸部"
		CASE "尿"
			STR:写真 += "用手指玩弄着尿道口"
		ENDSELECT
	ENDIF
;尿道
ELSEIF PREVCOM == 151
	STR:写真 += "将因为玩具而扩张開的尿道口"
	IF ABL:露出癖 >= 5
		STR:写真 += "展示给你看の"
	ELSE
		STR:写真 += "裸露出来"
	ENDIF
;展示肉棒
ELSEIF PREVPLAY("展示肉棒")
	PRINTFORM 你转向%CALLNAME:TARGET%挺出阴茎的同时
	STR:写真 += ""
	IF TALENT:淫乱 && BASE:欲求不満 >= 50 && RAND:3 == 0
		IF RAND:2
			STR:写真 += @"%TEXTR("用眼中浮现出爱心符号的发情的眼神/快要流出口水的恍惚表情")%持续不断地盯着你的阴茎的"
		ELSE
			STR:写真 += "温柔地握住你的阴茎的"
		ENDIF
	ELSEIF TALENT:好奇心 && RAND:2
		STR:写真 += "用想要吃进去试试看的眼神盯着你的阴茎的"
	ELSEIF (TALENT:開朗 || TALENT:喜歡引人注目) && RAND:2
		STR:写真 += "一边用一只手抚摸着你的阴茎一边做出姿势的"
	ELSEIF TALENT:害羞 || TALENT:清楚  && RAND:2
		STR:写真 += "在你的阴茎前露出了害羞的表情的"
	ELSEIF TALENT:叛逆
		STR:写真 += @"将你的阴茎推到一边%TEXTR("露出了不满的表情的/蔑视着这边的")%"
	ELSEIF TALENT:感情缺乏
		STR:写真 += "面无表情地看着你的阴茎的"
	ELSE
		STR:写真 += @"将你的阴茎推開%TEXTR("脸上泛起红潮的/微微飘逸着视线的")%"
	ENDIF
;其他
ELSE
	PRINTFORM 对准神情恍惚的吐出絮亂的氣息的%CALLNAME:TARGET%
	;ダブル剪刀手とかくぱぁとか
	IF MARK:屈服刻印 >= 2 && COND("指の使用") && TALENT:好色 + TALENT:淫乱 + COND("陶酔") + COND("満足") && RAND:2
		IF PREVPLAY("Ａ挿入")
			IF ABL:Ａ拡張 >= 3 && PREVPLAY("Ａ挿入中", PLAYER) && SIZE("阴茎", PLAYER) >= 2
				IF PREVEX("Ａ中出し")
					STR:写真 += "不檢點的精液垂流着的"
				ELSE
					STR:写真 += "肛門性交後中规中矩地张開的"
				ENDIF
			ELSEIF PREVEX("Ａ中出し")
				STR:写真 += "不斷流淌出%SPERMS%精液的"
			ELSE
				STR:写真 += "剛剛被充分侵犯過的"
			ENDIF
			STR:写真 += "尻穴"
		ELSEIF PREVPLAY("挿入")
			IF PREVEX("尿道内射精")
				STR:写真 += "不斷流淌出%SPERMS%精液的"
			ELSEIF PREVEX("Ｖ中出し")
				IF PREVEX("絶頂漏尿")
					STR:写真 += @"%TEXTR(@"漏尿之后从膣口逆流出精液的/刚刚絶頂放尿垂落出%SPERMS%精液的/因为自己尿液的残液濡湿而微微发光刚被中出的")%"
				ELSE
					STR:写真 += "不斷流淌出%SPERMS%精液的"
				ENDIF
			ELSEIF PREVEX("絶頂漏尿")
				STR:写真 += @"%TEXTR("刚刚漏出来还散发着气味的/刚刚激烈地喷溅出尿液的/被自己尿液的残液濡湿而微微发光")%"
			ELSE
				STR:写真 += "剛剛被充分侵犯過的"
			ENDIF
			IF PREVPLAY("Ｖ挿入")
				SIF COND("陰毛") && COND("陰毛") <= 3 && PREVEX("絶頂漏尿") == 0
					STR:写真 += "光滑的"
				STR:写真 += "阴部"
			ELSE
				STR:写真 += "尿道"
			ENDIF
		;女你用
		ELSE
			STR:写真 += @"%TEXTR("不斷激烈地痙攣著的/被愛液變得黏黏糊糊的")%阴部"
		ENDIF
		;Ｗ剪刀手乗り気
		IF TALENT:開朗 + TALENT:好色 + TALENT:淫乱 && RAND:3
			STR:写真 += "以Ｗ字開腿的姿勢主動地暴露着の"
		;Ｗ剪刀手陶酔
		ELSEIF COND("陶酔") && COND("大満足") && RAND:3
			STR:写真 += "露出的同时、带着荡漾着快楽的表情不检点地保持着Ｗ字開腿的姿勢の"
		ELSE
			STR:写真 += "用手指扩開、暴露在摄像机前の"
		ENDIF
	ELSE
		IF PREVPLAY("騎乗位")
			IF (TALENT:小悪魔 || TALENT:好色 || TALENT:淫乱) && RAND:2
				STR:写真 += @"%CALLNAME:PLAYER%を跨ぐように立ち、"
				IF PREVEX("Ｖ中出し") && PREVEX("Ａ中出し")
					STR:写真 += "精液垂流的前穴和後穴"
				ELSEIF PREVEX("Ｖ中出し")
					IF PREVEX("絶頂漏尿")
						STR:写真 += "尿和精液一并垂流的股間"
					ELSE
						STR:写真 += "%SPERMS%精液垂流的膣穴"
					ENDIF
				ELSEIF PREVEX("Ａ中出し")
					STR:写真 += "%SPERMS%精液垂流的肛門"
				ELSE
					STR:写真 += "剛剛被充分侵犯過的"
					IF PREVPLAY("Ｖ挿入")
						STR:写真 += "膣穴"
					ELSEIF PREVPLAY("Ａ挿入")
						STR:写真 += "肛門"
					ELSE
						STR:写真 += "尿道"
					ENDIF
				ENDIF
				STR:写真 += "对准镜頭展示の"
			ELSE
				STR:写真 += @"%CALLNAME:PLAYER%に跨ったまま"
				IF (TALENT:小悪魔 || TALENT:好色 || TALENT:淫乱) && RAND:2
					STR:写真 += "帶著妖艷的表情"
				ELSE
					STR:写真 += "带着佈滿紅潮的臉頰"
				ENDIF
				STR:写真 += "眼神恍惚地俯視著の"
			ENDIF
		;うつ伏せor屈曲位
		ELSEIF PREVPLAY("後背位") || PREVPLAY("屈曲位")
			IF PREVPLAY("後背位")
				IF RAND:2 && NAMES("床") == "床"
					STR:写真 += @"將床單緊緊抓握著、因短暫的脫力而伏在%NAMES("床")%上の"
				ELSE
					STR:写真 += @"渾身癱軟無力的趴伏在%NAMES("床")%上、短暫地休息著の"
				ENDIF
			ELSEIF PREVPLAY("屈曲位")
				STR:写真 += "保持着屈曲体勢の"
			ENDIF
			IF PREVEX("Ｖ中出し") && PREVEX("Ａ中出し")
				STR:写真 += "前後穴都有精液不斷流淌の"
			ELSEIF PREVEX("Ｖ中出し")
				IF PREVEX("絶頂漏尿")
					STR:写真 += "股間有尿和精液不斷流淌の"
				ELSE
					STR:写真 += @"从膣穴不断漏出%SPERMS%精液的"
				ENDIF
			ELSEIF PREVEX("Ａ中出し")
				IF PREVEX("前列腺高潮射精")
					STR:写真 += "肛門和阴茎中不断漏出精液的"
				ELSE
					STR:写真 += @"从肛門不断漏出%SPERMS%精液的"
				ENDIF
			ELSE
				STR:写真 += "精疲力尽の"
			ENDIF
		ELSEIF PREVPLAY("背面座位")
			IF PREVEX("前列腺高潮射精")
				STR:写真 += @"靠在%CALLNAME:PLAYER%的身上、从阴茎中咕噜咕噜地不断流出精液的"
			ELSE
				STR:写真 += @"因疲憊而倚靠在%CALLNAME:PLAYER%的懷中休息著の"
			ENDIF
		ELSE
			IF RAND:2 && NAMES("床") == "床"
				STR:写真 += "在被各種汁液浸透的床單上半倚著直起腰"
			ELSE
				STR:写真 += @"在%NAMES("床")%上仰面躺下"
			ENDIF
			IF PREVEX("Ｖ中出し") && PREVEX("Ａ中出し")
				STR:写真 += "、前後穴都有精液不斷流淌、"
			ELSEIF PREVEX("Ｖ中出し")
				IF PREVEX("絶頂漏尿")
					STR:写真 += "、阴部有尿和精液不斷流淌、"
				ELSE
					STR:写真 += "、膣穴有%SPERMS%精液不斷流淌、"
				ENDIF
			ELSEIF PREVEX("Ａ中出し")
				IF PREVEX("前列腺高潮射精")
					STR:写真 += "、尻穴和阴茎中不断流出精液、"
				ELSE
					STR:写真 += "、尻穴有%SPERMS%精液不斷流淌、"
				ENDIF
			ENDIF
			IF PREVEX("絶頂数合計") && RAND:4 == 0
				SIF TALENT:感情缺乏
					STR:写真 += "隐隐"
				STR:写真 += "用湿润的眼瞳看向这边的"
			ELSEIF TALENT:相思相愛 && RAND:2
				STR:写真 += "露出放荡的笑颜的"
			ELSEIF TALENT:不知羞恥 || TALENT:感情缺乏
				STR:写真 += "以舒緩的表情靜靜地發呆の"
			ELSE
				STR:写真 += @"用%TEXTR("両手/单手/手腕")%遮住眼睛休息著の"
			ENDIF
		ENDIF
	ENDIF
ENDIF

STR:写真 += @"%PHOTO_COSTUME(TARGET)%%CALLNAME:TARGET%"

STR:補足 += @"_%CALLNAME:TARGET%"
STR:補足 += "_事後"

SIF CHECK_CLO("泳装姿")
	STR:補足 += "_泳装"
SIF CHECK_CLO("体操服姿")
	STR:補足 += "_体操服"
SIF CHECK_CLO("学生服姿")
	STR:補足 += "_学生服"

;日期の記録
STR:補足 += @"_日期%TOSTR(DAY*10 + TIME)%"

SAVESTR:写真 = %STR:写真%

CALL SAVE_PHOTO, @"%STR:写真%%STR:補足%"

PRINTFORMW 摆好手机
PRINTFORML 咔嚓（攝像聲）
PRINTFORMW %CALLNAME:PLAYER%得到了『%SAVESTR:写真%』的写真。

;履歴を残す
VARSET LOCALS
SPLIT SAVESTR:Ｈ中の撮影履歴, "/", LOCALS

FOR LCOUNT, 0, 100
	SIF LOCALS:LCOUNT == ""
		BREAK
	;すでに似た写真があるかを格子
	SIF CHECK_HIGHMATCH(SAVESTR:写真, LOCALS:LCOUNT) == 0
		CONTINUE
	SAVESTR:写真 =
NEXT
SIF SAVESTR:写真 != ""
	SAVESTR:Ｈ中の撮影履歴 += @"%SAVESTR:写真%\/"

STR:青写真 =

@FLAG_COM158, ARG

@SOURCE_COM158, ARG
;とりあえず攝像機を少し弱めた感じ
CALL VARSET_COMF

SOURCE:0:露出 = 2000
SOURCE:0:逸脱 = 1000

;裸体
IF CHECK_CLO("裸体")
	;種族衣装が裸体でない
	IF CLO_ORI(TARGET, "服", "贴身制服", "外套", "胖次")
		SOURCE:0:露出 += 3000
		SOURCE:0:逸脱 += 1000
	ENDIF
;胖次のみ
ELSEIF CHECK_CLO("内衣姿")
	;種族衣装が服or贴身制服or外套を含む
	IF CLO_ORI(TARGET, "服", "贴身制服", "外套")
		SOURCE:0:露出 += 2000
		SOURCE:0:逸脱 += 500
	ENDIF
;没穿内褲
ELSEIF CHECK_CLO("没穿内褲")
	;種族衣装に内衣有
	IF CLO_ORI(TARGET, "胖次")
		SOURCE:0:露出 += 1000
		SOURCE:0:逸脱 += 300
	ENDIF
ENDIF

;拘束の状態により露出を追加
IF GETBIT(TEQUIP:拘束, 5)
	SOURCE:0:露出 += 1000
ELSEIF GETBIT(TEQUIP:拘束, 4)
	SOURCE:0:露出 += 500
ELSEIF GETBIT(TEQUIP:拘束, 3)
	SOURCE:0:露出 += 200
ENDIF

;露出はTRACHECKで補正がかかるので、逸脱に害羞などで補正
SIF TALENT:害羞
	TIMES SOURCE:0:逸脱, 2.00
SIF TALENT:清楚
	TIMES SOURCE:0:逸脱, 1.50
SIF TALENT:喜歡引人注目
	TIMES SOURCE:0:逸脱, 0.50
SIF TALENT:内気
	TIMES SOURCE:0:逸脱, 1.70
SIF TALENT:聰慧
	TIMES SOURCE:0:露出, 1.50
SIF TALENT:笨蛋
	TIMES SOURCE:0:露出, 0.70

SOURCE:0:恐惧 += SOURCE:0:逸脱 / 10

CALL CALC, "調教ソース", ARG


code:
@COM15
;尿道指挿入
;指はブジーに比べると非常に入れ辛いので、拡張効果があるような段階では入れることは出来ない
;あくまで尿道の広がっている相手への愛撫指令。消耗は低い
CALL PRINT_TRAIN_NAME(SELECTCOM)

CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

RETURN 1

@FLAG_COM15, ARG
CALL SETFLAG, "尿道性感フラグ", TARGET
;尿的污垢発生
IF PENIS(TARGET)
	STAIN:Ｐ = SET_STAIN(STAIN:Ｐ, "尿")
ELSE
	STAIN:Ｖ = SET_STAIN(STAIN:Ｖ, "尿")
ENDIF
STAIN:PLAYER:手 = SET_STAIN(STAIN:PLAYER:手, "尿")
;胖次をずらす
SIF !CHECK_CLO("Ｖ露出") && INSERT("挿入") == 0
	CALL SETFLAG, "ボトムを露出：テ接吻ト没有", TARGET
SIF TEQUIP:尿道振動棒
	CALL FLAG_COM152, 130

@SOURCE_COM15, ARG
IF TEQUIP:尿道振動棒
	CALL SOURCE_INSERTION, "尿道振動棒", ARG, 130
ELSE
	CALL SOURCE_INSERTION, "尿道指挿入", ARG, 100
ENDIF

@TRAIN_MESSAGE_COM15
PRINTFORM %CALLNAME:PLAYER%将
IF TEQUIP:接吻
	PRINTFORM %TEXTS("接吻継続")%、
	CALL SETFLAG, "追加接吻フラグ", TARGET
ELSEIF TEQUIP:六九式
	PRINTFORM %TEXTS("６９")%
ELSEIF INSERT("Ａ挿入")
	PRINTFORM 侵犯著肛門的同時、
ELSEIF PREVCOM == 1 && PENIS(TARGET) == 0 && CHECK_CLO("Ｖ露出")
	SIF CALCF("Ｃ発情")
		PRINT 興奮勃起的
	PRINTFORM 陰舔舐著蒂仔細的同時、
ELSEIF PREVCOM == 13 && PENIS(TARGET) == 0 && TEQUIP:尿道振動棒
	PRINT 觀賞被手指撥開的陰部的同時、
ENDIF
PRINTFORM %CALLNAME:TARGET%的

IF TEQUIP:尿道振動棒
	IF COND("尿道布ル")
		PRINTFORM 膀胱用一顆顆的尿道拉珠慢慢填滿、將尿道口因異常的快樂而不斷張合的下流姿態盡情的享受著……
	ELSEIF SIZE("尿道振動棒", TARGET) >= 2
		PRINTFORM 將頂著膀胱不斷壓迫著腹部的內臟的%TEXTR("粗大/大型/兇惡")%振動棒
		SIF ABL:PLAYER:施虐属性 >= 3
			PRINTFORM %TEXTR("粗暴地/毫無顧忌/強硬地/肆無忌憚地/強行地")%
		PRINTFORMW 抽插著……
	ELSEIF RAND:3
		PRINTFORMW 將在尿道的內部不斷激烈震動著的振動棒前後抽插著……
	ELSEIF RAND:2
		PRINTFORMW 將牢牢固定在尿道中的振動棒反復抽插著、氣勢十足的衝撞著頂端的肉襞……
	ELSE
		PRINTFORMW 將牢牢固定在尿道中的振動棒來迴旋轉著、隔著膀胱的內壁直接壓迫著腹部的臟器……
	ENDIF

	PRINTFORM %CALLNAME:TARGET%
	IF TALENT:尿道狂
		IF RAND:3 == 0
			PRINTFORMW 完全化作尿道快感的俘虜、雙眼上翻而雙唇大張、用放蕩而下流的表情享受著異常的快感……
		ELSEIF RAND:2 == 0
			PRINTFORMW 因過強的快感刺激而無意識的顫抖著四肢、全心地沉迷在異常的快樂之中……
		ELSE
			PRINTFORMW 即使尿道、膀胱乃至於腹腔之中的臟器都在劇烈的震動下變得亂七八糟、但卻將這份痛苦視作同樣的快樂、將其從內心深處的享受著……
		ENDIF
	ELSEIF TALENT:尿道性感
		IF RAND:3 == 0
			PRINTFORMW 在劇烈的快感之下變得神志恍惚、將%CALLNAME:PLAYER%所教導的異常快樂盡情的享受著……
		ELSEIF RAND:2 == 0
			PRINTFORMW 用被教育成性器官的尿道將異常的快感全數接下、盡情的享受著……
		ELSE
			PRINTFORMW 將尿道所傳遞的快樂用身軀深切地體會著、在無法抑制地顫抖起來的同時、也不時漏出小聲的嬌喘……
		ENDIF
	ELSEIF CHECK("尿道PLAY") == 0
		IF RAND:3 == 0
			PRINTFORMW 以苦悶的表情將不合理的疼痛感忍耐著……
		ELSEIF RAND:2 == 0
			PRINTFORMW 咬緊牙關、將過激的異常調教忍耐著……
		ELSE
			PRINTFORMW 緊皺著眉頭、零星地灑落淚滴的同時、拼命地忍耐著痛苦……
		ENDIF
	ELSE
		IF RAND:3 == 0
			PRINTFORMW 因尿道所傳來的異常快感而迷惑著……
		ELSEIF RAND:2 == 0
			PRINTFORMW 在%CALLNAME:PLAYER%所給予的異常感中迷惑著……
		ELSE
			PRINTFORMW 似乎對於尿道能變成產生快感的地方這件事、難以掩飾心中的迷惑……
		ENDIF
	ENDIF
ELSEIF PREVCOM == SELECTCOM
	IF RAND:2 && PENIS(TARGET) == 0
		PRINTFORM 膀胱用指尖插入、輕輕彎曲的同時
		IF RAND:2
			PRINTFORMW 用修剪乾淨的指甲小心的輕輕摩擦著內部的粘膜……
		ELSE
			PRINTFORMW 將內尿道口用指腹集中地摩挲著……
		ENDIF
	ELSE
		IF RAND:2
			PRINTFORMW 尿道用指尖插入直到根部、輕柔而緩慢的進出抽插地持續刺激著……
		ELSE
			PRINTFORMW 尿道的敏感粘膜用手指溫柔地摩挲著……
		ENDIF
	ENDIF
;指
ELSE
	;前回尿道で射精された
	IF PREVEX("尿道内射精")
		PRINTFORM 被強迫灌入濃郁精液的
	;前回放尿
	ELSEIF PREVPLAY("放尿")
		PRINTFORM 剛剛排泄出晶瑩尿液的
	ELSEIF PREVCOM == 83
		PRINTFORM 充滿了性愛之後的餘味的
	ELSEIF PREVCOM == 151
		PRINTFORM 被尿道探針蹂躪之後的
	ENDIF
	PRINTFORM \@ PENIS(TARGET) ? 陰莖鈴口 # 尿道 \@用
	IF CALCF("尿道径") > SIZE("指の直径", PLAYER)*2 && RAND:2
		PRINTFORM \@ RAND:2 ? 中指和薬指 # 兩根手指 \@
	ELSE
		PRINTFORM \@ RAND:2 ? 中 # \@指
	ENDIF
	PRINTFORM \@ RAND:2 ? 直達根部地 # 輕柔緩慢地 \@挿入體內

	IF RAND:3 == 0 && BASE:尿意 >= 10000 && PENIS(TARGET) == 0
		PRINTFORMW 隨後、享受著內部積存的溫暖尿液。
	ELSEIF RAND:2
		PRINTFORMW 隨後、一邊小心地注意不傷到粘膜、一邊用指尖輕輕描畫著。
	ELSE
		PRINTFORMW 隨後、用指腹輕輕摩挲著吸附過來的粘膜。
	ENDIF

	;初めての指受け入れ
	IF FIRSTCOM(SELECTCOM) == 0
		IF TALENT:尿道性感
			PRINTFORMW 為尿道性感所俘獲的%CALLNAME:TARGET%難耐地扭動著腰肢、用嬌媚的語氣催促著更加激烈的愛撫……
		ELSE
			PRINTFORMW %CALLNAME:TARGET%似乎相當不安的樣子、在承受著並帶著些許的反抗迎接手指的活動……
		ENDIF
	;前回中出し
	ELSEIF PREVEX("尿道内射精")
		PRINTFORM 從尿道中拔出的手指沾滿渾濁的精液、看見這淫靡的景色之後、
		IF ABL:精液中毒 > 3 && PALAM:欲情 >= 10000
			PRINTFORMW %CALLNAME:TARGET%像是十分渴望一樣的舔舐著唇角……
		ELSEIF TALENT:恋慕
			PRINTFORMW %CALLNAME:TARGET%用恍惚而快樂的眼神注視著它……
		ELSE
			IF RAND:2
				PRINTFORMW %CALLNAME:TARGET%的臉頰因羞澀染上紅潮……
			ELSE
				PRINTFORMW %CALLNAME:TARGET%害羞的偏開視線……
			ENDIF
		ENDIF
	ELSEIF TALENT:尿道性感 && CHECK("尿道PLAY")
		SELECTCASE EXP:尿道経験
		CASE IS <= 50
			PRINTFORMW %CALLNAME:TARGET%開始嘗試享受手指抽插的觸覺……
		CASE IS <= 80
			PRINTFORMW %CALLNAME:TARGET%恍惚地感受著手指抽動的觸覺與快樂……
		CASE IS <= 120
			PRINTFORMW %CALLNAME:TARGET%因私密場所被輕輕翻攪搔弄而產生的奇妙快樂而苦悶的漏出了灼熱的喘息……
		CASEELSE
			PRINTFORMW %CALLNAME:TARGET%無意識的流下涎水、貪圖快樂地揮舞著腰肢套弄著手指……
		ENDSELECT
	ELSE
		PRINTFORMW %CALLNAME:TARGET%調節著絮亂的呼吸、漲紅著臉忍耐著在尿道中進出的手指……
	ENDIF
ENDIF


code:
@COM170
;攝像機
;害羞や清楚、贞操重视持ちだと凄い反発。
CALL PRINT_TRAIN_NAME(SELECTCOM)

CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

IF TEQUIP:攝像機
	STR:前回指令 = 攝像機(撮影開始)
ELSE
	STR:前回指令 = 攝像機(结束摄影)
ENDIF

RETURN 1

@TRAIN_MESSAGE_COM170
;视频拍摄がされていれば終了させる
IF TEQUIP:攝像機
	;短すぎる
	IF TEQUIP:攝像機 <= 3
		;認識阻害持ち
		IF TALENT:MASTER:禁断的知識 >= 3
			PRINTFORMW 如果现在停止拍摄的东西就太少了、要停止摄影吗？
		ELSE
			PRINTFORMW 如果现在停止拍摄的东西就太少了、没有保留的价值、要停止摄影吗？
		ENDIF
		CALL PRINT_SELECT, "是/否"
		IF RESULT
			TFLAG:指令中止 = 1
			RETURN 0
		ENDIF
		STR:ビデオ内容 =
		STR:ビデオジャンル =
	ENDIF

	TEQUIP:攝像機 = 0
	PRINTW ★★★視頻攝影結束★★★
	;助手が居ない
	IF ASSI <= 0 || ASSIPLAY

	ELSEIF ABL:ASSI:自慰中毒 >= RAND:6 && RAND:2
		IF TALENT:ASSI:冷静 + TALENT:ASSI:感情缺乏 && RAND:2
			PRINTFORMW 攝影結束後的%CALLNAME:ASSI%盡力保持著冷靜、但仍然無法抑制的伸手愛撫著因渴求快樂而發熱的股間……
		ELSE
			PRINTFORMW 在充分的觀賞過兩人的淫戲後、%CALLNAME:ASSI%無法抑制體內的情慾、向濡濕的股間伸出了手……
		ENDIF
	ELSE
		IF TALENT:ASSI:容易濕 && ABL:ASSI:欲望 >= RAND:6 && RAND:2
			PRINTFORMW %CALLNAME:ASSI%作為%TEXTS("助手の名称")%良好的完成了委託、但淫液從股間溢出、在大腿上畫出了淫靡的線條……
		ELSE
			PRINTFORMW %CALLNAME:ASSI%拍攝到十分良好的畫面、因此而開心的笑著……
		ENDIF
	ENDIF
ELSE
	IF STR:ビデオ内容 != ""
		PRINTFORMW 现在已经有収録的部分了、要放棄然后重新拍摄吗？
		CALL PRINT_SELECT, "是/否"
		IF RESULT
			TFLAG:指令中止 = 1
			RETURN 0
		ENDIF
	ENDIF

	TEQUIP:攝像機 = 1
	STR:ビデオ内容 =
	STR:ビデオジャンル =
	PRINTFORML ★★★視頻拍攝開始★★★

	IF EXP:撮影経験 == 0
		IF TALENT:好奇心
			PRINT 將對初次見到的機器興趣滿滿的樣子的
		ELSE
			PRINT 將像在看不可思議的東西一樣凝視著撮影機材的
		ENDIF
	ELSEIF ABL:露出癖 >= 3 || TALENT:不知羞恥
		IF TALENT:相思相愛 && TALENT:淫乱 && RAND:2
			PRINT 將為了向他人展示自己的愛而更加熱烈了的
		;ELSEIF COND("人気：ＡＶ") && RAND:2
		;	IF TALENT:強気 || TALENT:高慢
		;		PRINT 自分の人気の有些様を受けて満更でもない様子で準備をしている
		;	ELSE
		;		PRINT 作品作りにも慣れてレンズ越しのお客からの受けを思案している
		;	ENDIF
		ELSE
			PRINT 將在攝像機的鏡頭瞄準後、突然的、雙眼閃耀起來了的
		ENDIF
	ELSEIF TEQUIP:野外PLAY
		PRINT 將緊張不安的環伺著周圍的情況的
	ELSEIF TALENT:害羞
		PRINT 將臉頰通紅的偷偷窺視著這邊的情況的
	ELSE
		PRINT 將在认真地考虑着要进行什么様的PLAY的
	ENDIF
	PRINTFORM %CALLNAME:TARGET%斜視著、
	IF ASSI <= 0
		PRINTFORMW %CALLNAME:PLAYER%將攝影機小心地平穩放置在合適的高台上、重新開始了。
		PRINTFORMW （介于%TEXTS("助手の名称")%並不在場、拍攝的質量可能會有跌落、但這也是沒有辦法的事……）
	ELSEIF ASSIPLAY
		PRINTFORMW %CALLNAME:PLAYER%將攝影工作暫時給%CALLNAME:MASTER%之後、重新開始了。
	ELSE
		LOCALS =
		IF TEQUIP:野外PLAY || TEQUIP:室内露出PLAY
			IF CHARA_NUM("室内露出観客数")
				LOCALS = 和在場的少女們一起竊笑的同時、
			ELSEIF ABL:ASSI:施虐属性 >= 3
				LOCALS = 惋惜少女們並不在場的同時、
			ELSE
				LOCALS = 確認週邊沒有人在場的同時、
			ENDIF
		ELSEIF TEQUIP:室内露出PLAY
			LOCALS = 部員たちの反応も一緒に記録することを意識し的同时、
		ELSEIF TALENT:ASSI:母親 + TALENT:ASSI:父親 && RAND:3 == 0
			LOCALS = 心中想著要將我的孩子的性的成長記錄下來的同時、
		ELSEIF TALENT:母親 + TALENT:父親 && COND("兄弟", ASSI) && RAND:3 == 0
			LOCALS = 對將自己的親人的情事記錄下來而興奮的同時、
		ELSEIF (TALENT:ASSI:淫乱 || TALENT:ASSI:相思相愛) && RAND:4 == 0
			IF TALENT:ASSI:淫乱 && RANDIF(1, TALENT:ASSI:相思相愛)
				IF TALENT:ASSI:高慢
					LOCALS = 發出製造過激場面的指示的同時、
				ELSE
					LOCALS = 對之後會出現的情事飽含期待的同時、
				ENDIF
			ELSE
				IF TALENT:ASSI:淫乱
					LOCALS = 「現在是%CALLNAME:TARGET%的時間」這樣自言自語的同時、
				ELSE
					LOCALS = 浮現出寂寞表情但還是勉強振作的同時、
				ENDIF
			ENDIF
		ELSEIF ABL:ASSI:自慰中毒 >= 3 && RAND:3 == 0
			LOCALS = 一隻手不自覺的向\@ TALENT:ASSI:男性 ? 股間 # 秘所 \@伸出、尽力忍耐的同時、

		ELSEIF ABL:ASSI:施虐属性 >= 3 && RANDIF(1, TALENT:ASSI:坦率, TALENT:ASSI:好奇心, TALENT:ASSI:鼓舞, TALENT:ASSI:粗野)
			LOCALS = 在懊惱著不能直接蹂躪對方並努力的工作著的同時、
		ELSEIF TALENT:ASSI:粗野 && RANDIF(1, TALENT:ASSI:坦率, TALENT:ASSI:好奇心, TALENT:ASSI:鼓舞)
			LOCALS = 因接下委託而挺起胸給自己鼓勁的同時、
		ELSEIF TALENT:ASSI:鼓舞 && RANDIF(1, TALENT:ASSI:坦率, TALENT:ASSI:好奇心)
			LOCALS = 為%CALLNAME:TARGET%活躍地應援著的同時、
		ELSEIF TALENT:ASSI:好奇心 && RANDIF(1, TALENT:ASSI:坦率)
			IF RELATION:ASSI:TARGET >= 150
				LOCALS = 為了不讓%CALLNAME:TARGET%的痴態白白浪費而拼命的記錄著的同時、
			ELSE
				LOCALS = 為了不讓%CALLNAME:MASTER%們的痴態白白浪費而拼命的記錄著的同時、
			ENDIF
		ELSEIF TALENT:ASSI:坦率
			LOCALS = 仔細地聽從著%CALLNAME:MASTER%命令的同時、
		ELSEIF ABL:ASSI:自慰中毒 >= 3 && RAND:3 == 0
			LOCALS = 一隻手不自覺的向\@ TALENT:ASSI:男性 ? 股間 # 秘所 \@伸出、尽力忍耐的同時、
		ENDIF

		IF LOCALS == ""
			PRINTFORMW %CALLNAME:PLAYER%將攝像機遞交給%CALLNAME:ASSI%、開始攝影了……
		ELSE
			PRINTFORMW %CALLNAME:PLAYER%將攝像機遞交給%CALLNAME:ASSI%。
			PRINTFORMW %CALLNAME:ASSI%%LOCALS%開始攝影了……
		ENDIF
		CALL KOJO_GUEST, "助手：撮影開始", ASSI
	ENDIF
ENDIF

@EQUIP_COM170
SIF TEQUIP:攝像機 == 0
	RETURN 0
SIF SELECTCOM == 170
	RETURN 0

CALL SOURCE_COM170, 10

RETURN 1

@FLAG_COM170, ARG

@SOURCE_COM170, ARG
SIF TEQUIP:攝像機 == 0
	RETURN 0

CALL VARSET_COMF

SOURCE:0:露出 = 3000
SOURCE:0:逸脱 = 5000

;裸体
IF CHECK_CLO("裸体")
	;種族衣装が裸体でない
	IF CLO_ORI(TARGET, "服", "贴身制服", "外套", "胖次")
		SOURCE:0:露出 += 3000
		SOURCE:0:逸脱 += 1000
	ENDIF
;胖次のみ
ELSEIF CHECK_CLO("内衣姿")
	;種族衣装が服or贴身制服or外套を含む
	IF CLO_ORI(TARGET, "服", "贴身制服", "外套")
		SOURCE:0:露出 += 2000
		SOURCE:0:逸脱 += 500
	ENDIF
;没穿内褲
ELSEIF CHECK_CLO("没穿内褲")
	;種族衣装に内衣有
	IF CLO_ORI(TARGET, "胖次")
		SOURCE:0:露出 += 1000
		SOURCE:0:逸脱 += 300
	ENDIF
ENDIF

;拘束の状態により露出を追加
IF GETBIT(TEQUIP:拘束, 5)
	SOURCE:0:露出 += 1000
ELSEIF GETBIT(TEQUIP:拘束, 4)
	SOURCE:0:露出 += 500
ELSEIF GETBIT(TEQUIP:拘束, 3)
	SOURCE:0:露出 += 200
ENDIF

;露出はTRACHECKで補正がかかるので、逸脱に害羞などで補正
SIF TALENT:害羞
	TIMES SOURCE:0:逸脱, 2.00
SIF TALENT:清楚
	TIMES SOURCE:0:逸脱, 1.50
SIF TALENT:喜歡引人注目
	TIMES SOURCE:0:逸脱, 0.50
SIF TALENT:内気
	TIMES SOURCE:0:逸脱, 1.70
SIF TALENT:聰慧
	TIMES SOURCE:0:露出, 1.50
SIF TALENT:笨蛋
	TIMES SOURCE:0:露出, 0.70

SOURCE:0:恐惧 += SOURCE:0:逸脱 / 10

CALL CALC, "調教ソース", ARG

@EQUIP_COM170_2
SIF TEQUIP:攝像機 == 0
	RETURN 0
SIF SELECTCOM == 170
	RETURN 0

PRINTFORML ＜视频拍摄中{TEQUIP:攝像機}/10＞
EXP:露出経験 += 1
EXP:撮影経験 += 1

RETURN 1

;ターン終了時の色々な格子
@EQUIP_COM170_EVENTCOMEND
;ここでは主にジャンルの格子をする
;撮影開始時
IF TEQUIP:攝像機 == 1
	IF CHECK_SHIRT("女教師")
		STR:ビデオジャンル += "女教師/"
	ELSEIF CHECK_SHIRT("女仆装")
		STR:ビデオジャンル += "女仆/"
	ELSEIF CHECK_OUTER("布鲁马")
		STR:ビデオジャンル += "布鲁马/"
	ELSEIF CHECK_SHIRT("体操服")
		STR:ビデオジャンル += "体操服/"
	ELSEIF CHECK_SHIRT("看護婦")
		STR:ビデオジャンル += "护士/"
	ELSEIF CHECK_SHIRT("美式校服")
		STR:ビデオジャンル += "美式校服/"
	ELSEIF CHECK_SHIRT("水手服")
		STR:ビデオジャンル += "水手服/"
	ELSEIF CHECK_SHIRT("学生服")
		STR:ビデオジャンル += "学生服/"
	ELSEIF CHECK_SHIRT("巫女")
		STR:ビデオジャンル += "巫女/"
	ELSEIF CHECK_CLO("泳装姿")
		IF CHECK_CLO("Micro比基尼姿")
			STR:ビデオジャンル += "Micro比基尼/"
		ELSEIF CHECK_CLO("比基尼姿")
			STR:ビデオジャンル += "比基尼/"
		ELSEIF CHECK_BODYS("競泳泳装")
			STR:ビデオジャンル += "競泳泳装/"
		ELSEIF CHECK_BODYS("死库水")
			STR:ビデオジャンル += "死库水/"
		ELSE
			STR:ビデオジャンル += "泳装/"
		ENDIF
	ELSEIF CHECK_CLO("拘束衣")
		STR:ビデオジャンル += "拘束衣/"
	ELSEIF CHECK_BODYS("兔")
		STR:ビデオジャンル += "兔/"
	ELSEIF CHECK_SHIRT("哥特蘿莉礼服")
		STR:ビデオジャンル += "ゴス萝莉/"
	ELSEIF CHECK_SHIRT("啦啦队制服")
		STR:ビデオジャンル += "チアガール/"
	ELSEIF CHECK_SHIRT("旗袍")
		STR:ビデオジャンル += "旗袍/"
	ENDIF

	SIF TALENT:男性 && TALENT:PLAYER:男性
		STR:ビデオジャンル += "ＢＬ/"
	SIF TALENT:女性 && TALENT:PLAYER:女性
		STR:ビデオジャンル += "レズ/"
	SIF TALENT:扶她 || TALENT:PLAYER:扶她
		STR:ビデオジャンル += "扶她/"
	SIF TEQUIP:野外PLAY
		STR:ビデオジャンル += "野外/"

	IF TALENT:巨乳 >= 3
		STR:ビデオジャンル += "超乳/"
	ELSEIF TALENT:巨乳 >= 2
		STR:ビデオジャンル += "爆乳/"
	ELSEIF TALENT:巨乳
		STR:ビデオジャンル += "巨乳/"
	ELSEIF TALENT:貧乳
		STR:ビデオジャンル += "貧乳/"
	ENDIF

	SIF TALENT:巨尻
		STR:ビデオジャンル += "巨尻/"

	SELECTCASE COND("陰毛")
	CASE 1 TO 3
		STR:ビデオジャンル += "白虎/"
	CASE IS >= 7
		STR:ビデオジャンル += "剛毛/"
	ENDSELECT

	SIF TALENT:魅力
		STR:ビデオジャンル += "美少女/"
ENDIF
SIF SELECTCOM == 170
	RETURN 0

;このターンのPLAYを記録
SIF SELECTCOM >= 0
	STR:ビデオ内容 += @"%TRAINNAME:SELECTCOM%\/"

SIF NOWEX:処女喪失
	STR:ビデオジャンル += "処女喪失/"
SIF NOWEX:噴乳 || COND("喝母乳フラグ", PLAYER)
	STR:ビデオジャンル += "母乳/"
SIF NOWEX:絶頂漏尿
	STR:ビデオジャンル += "放尿/"
SIF NOWEX:排泄
	STR:ビデオジャンル += "スカ/"
;SIF CHECK_ECST(TARGET) >= 4
;	STR:ビデオジャンル += "多重絶頂/"
SIF COND("清洁口交フラグ")
	STR:ビデオジャンル += "お掃除/"
SIF COND("Ａ清洁口交フラグ")
	STR:ビデオジャンル += "ＡＴＭ/"
SIF TEQUIP:眼罩 + TEQUIP:拘束 + TEQUIP:口枷
	STR:ビデオジャンル += "拘束/"
SIF COND("童貞喪失フラグ")
	STR:ビデオジャンル += "童貞喪失/"
SIF COND("童貞喪失フラグ", PLAYER)
	STR:ビデオジャンル += "童貞狩猎/"

;指令依存
SELECTCASE SELECTCOM
;使用前と使用後の比較
CASE TNUM("自慰")
	STR:ビデオジャンル += "自慰/"
CASE TNUM("尿道指挿入"), TNUM("尿道导管"), TNUM("尿道探针"), TNUM("尿道振動棒")
	STR:ビデオジャンル += "尿道/"
CASE TNUM("扒開阴道"), TNUM("扒開菊穴")
	STR:ビデオジャンル += "局部アップ/"
CASE TNUM("振動棒")
	SIF SIZE("振動棒") >= 3
		STR:ビデオジャンル += "拡張/"
CASE TNUM("肛用振動棒")
	STR:ビデオジャンル += "肛門/"
	SIF SIZE("Ａ振動棒") >= 3
		STR:ビデオジャンル += "拡張/"
CASE TNUM("正常位") TO TNUM("子宮挿入")
	STR:ビデオジャンル += "性交/"
CASE TNUM("肛門正常位") TO TNUM("肛門騎乗位")
	STR:ビデオジャンル += "肛門性交/"
CASE TNUM("尿道正常位") TO TNUM("尿道騎乗位")
	STR:ビデオジャンル += "尿道性交/"
CASE TNUM("強制口交")
	STR:ビデオジャンル += "強制口交/"
CASE TNUM("拳交")
	STR:ビデオジャンル += "拡張/拳交/"
CASE TNUM("肛門拳交") TO TNUM("両穴拳交")
	STR:ビデオジャンル += "拡張/拳交/肛門/"
ENDSELECT

SIF COND("飲尿フラグ", PLAYER)
	STR:ビデオジャンル += "聖水/"
SIF COND("飲尿フラグ", TARGET)
	STR:ビデオジャンル += "飲尿/"

IF TEQUIP:攝像機 >= 10
	TEQUIP:攝像機 = 0
	PRINTL
	PRINTL 拍这么多就足够了吧。
	PRINTW ＜视频拍摄終了＞
	RETURN 1
ELSEIF BASE:体力 < 100
	TEQUIP:攝像機 = 0
	PRINTFORML
	PRINTFORML 从%CALLNAME:TARGET%那边漂来了疲惫的气氛…
	PRINTFORMW 似乎停止撮影会比较好。
	PRINTW ＜视频拍摄終了＞
	RETURN 1
ENDIF

TEQUIP:攝像機 += 1

RETURN 1


code:
@COM171
;野外PLAY
;TEUIP:54のビットが 0人里, 1森林, 2花園, 3水辺, 4海
CALL PRINT_TRAIN_NAME(SELECTCOM)

CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

IF TEQUIP:野外PLAY == 0
	PREVCOM = 171
	RETURN 0
ELSEIF GETBIT(TFLAG:地の文分岐, 0)
	STR:前回指令 = 野外PLAY(人里)
ELSEIF GETBIT(TFLAG:地の文分岐, 1)
	STR:前回指令 = 野外PLAY(森林)
ELSEIF GETBIT(TFLAG:地の文分岐, 2)
	STR:前回指令 = 野外PLAY(花園)
ELSEIF GETBIT(TFLAG:地の文分岐, 3)
	STR:前回指令 = 野外PLAY(水辺)
ELSEIF GETBIT(TFLAG:地の文分岐, 4)
	STR:前回指令 = 野外PLAY(海)
ENDIF

RETURN 1

@TRAIN_MESSAGE_COM171
#DIM MEMO_KISEKAE
#DIM MEMO_LINECOUNT

IF TEQUIP:野外PLAY == 0
	PRINTFORML 那麼、要和%CALLNAME:TARGET%一起到哪裡去呢？
	PRINTFORML
	PRINTL  [ 0] 一起到聚居地去吧
	PRINTL  [ 1] 一起到小樹林去吧
	PRINTL  [ 2] 一起到花園去吧
	PRINTL  [ 3] 一起到水辺去吧
	PRINTL  [ 4] 一起到海辺去吧
	PRINTL [100] 不、再考慮一下吧。
	CALL INPUT_SELECT, 5, 100
ELSEIF TEQUIP:野外PLAY
	;野外PLAY行き先フラグ
	TFLAG:地の文分岐 = TEQUIP:野外PLAY
	PRINTFORMW 和%CALLNAME:TARGET%回到部屋了……
	TEQUIP:野外PLAY = 0
	SIF ASSI
		TEQUIP:ASSI:野外PLAY = 0
	CALL KOJO_MESSAGE_COM
	RETURN 1
ENDIF

IF RESULT == 100
	TFLAG:指令中止 = 1
	RETURN 0
ELSEIF RESULT == 0
	LOCALS = 人里
ELSEIF RESULT == 1
	LOCALS = 森林
ELSEIF RESULT == 2
	LOCALS = 花園
ELSEIF RESULT == 3
	LOCALS = 水辺
ELSEIF RESULT == 4
	LOCALS = 海
ENDIF

SETBIT TEQUIP:野外PLAY, RESULT
SIF ASSI
	TEQUIP:ASSI:野外PLAY = TEQUIP:野外PLAY

;野外PLAY行き先フラグ
TFLAG:地の文分岐 = TEQUIP:野外PLAY

;換装させた状態を格子
MEMO_KISEKAE = 0

;選択肢の存在判定に用いる
VARSET LOCAL
;選択肢の二重出現を防ぐ
MEMO_LINECOUNT = LINECOUNT

;其他の場合
IF MEMO_LINECOUNT == LINECOUNT && EQUIP:胖次
	PRINTFORML ……在外出之前、要先脫下%CALLNAME:TARGET%的內褲嗎？
	SETCOLOR DEF_COLOR("黄色")
	PRINTFORM 現在的服装：
	CALL PRINT_DRESSL, TARGET
	RESETCOLOR
	PRINTFORML
	PRINTFORML  [ 0] 是
	PRINTFORML  [ 1] 否

	CALL INPUT_SELECT, 2

	SELECTCASE RESULT
	CASE 0
		CALL SET_PANTIES, ""
		MEMO_KISEKAE = 3
	ENDSELECT
	PRINTL
ENDIF

IF MEMO_KISEKAE == 1
	PRINT 順從地更衣的
ELSEIF CHECK_CLO("裸体")
	PRINT 預先脫光的
ENDIF
PRINTFORM %CALLNAME:TARGET%
IF MEMO_KISEKAE == 2
	PRINTFORM 換上了%SHOES(TARGET, "略")%、
	IF CHECK_CLO("裸体")
		PRINT 現在只穿著這個、
	ELSE
		PRINT 現在穿著這個、
	ENDIF
ELSEIF MEMO_KISEKAE == 3 && CHECK_CLO("裸体") == 0
	PRINT 內褲被預先收走了、
ELSEIF MARK:屈服刻印 == 3 || TALENT:恋慕
	PRINT 相互牽著手、
ENDIF
PRINTFORMW 一起到%LOCALS%那裡去了……
PRINTFORM %CALLNAME:TARGET%
IF (ABL:露出癖 >= 3 || TALENT:害羞) && GETBIT(TEQUIP:野外PLAY, 0)
	PRINTFORMW 感覺到有什麼人在注視著、臉頰被紅潮席捲、身體微微的顫抖著……
ELSEIF ABL:露出癖 >= 3
	PRINTFORMW 想像著盯著自己的視線、臉頰被紅潮席捲、身體微微的顫抖著……
ELSEIF TALENT:害羞
	PRINTFORMW 對細微的聲響也會敏感的做出反應的身體在微弱的顫抖著……
ELSEIF TALENT:小悪魔 || TALENT:坏心眼
	PRINTFORM 在行走的同時、抱住%CALLNAME:MASTER%的手臂、一邊凝視著臉頰、一邊用
	IF TALENT:娇小
		PRINT 發育不全的
	ELSEIF TALENT:巨乳
		PRINT 豊満的
	ELSE
		PRINT 十分柔軟的
	ENDIF
	PRINTW 身體輕輕摩擦著……
ELSEIF TALENT:好奇心
	PRINTFORMW 對四周的風景都是一副十分好奇的樣子……
ELSE
	PRINTFORMW 似乎在想像著可能在前方發生的事情……
ENDIF

RETURN 1

@FLAG_COM171, ARG

@SOURCE_COM171, ARG
CALL VARSET_COMF

IF GETBIT(TEQUIP:野外PLAY, 0)
	SOURCE:0:露出 = 2000
	SOURCE:0:逸脱 = 400
ELSEIF GETBIT(TEQUIP:野外PLAY, 1)
	SOURCE:0:露出 = 400
ELSEIF GETBIT(TEQUIP:野外PLAY, 2)
	SOURCE:0:露出 = 400
ELSEIF GETBIT(TEQUIP:野外PLAY, 3) || GETBIT(TEQUIP:野外PLAY, 4)
	SOURCE:0:露出 = 600
ENDIF
SOURCE:0:逸脱 += 500

SOURCE:0:充足 += ABL:露出癖 * 300

;服の有無で羞恥、逸脱度が上昇。人里だと物凄い事に。
IF CHECK_CLO("裸体")
	;種族衣装が裸体でない
	IF CLO_ORI(TARGET, "服", "贴身制服", "外套", "胖次")
		IF GETBIT(TEQUIP:野外PLAY, 0)
			SOURCE:0:露出 += 6000
			SOURCE:0:逸脱 += 1000
		ELSE
			SOURCE:0:露出 += 2000
			SOURCE:0:逸脱 += 500
		ENDIF
	ENDIF
ELSEIF CHECK_CLO("内衣姿")
	;種族衣装が服or贴身制服or外套を含む
	IF CLO_ORI(TARGET, "服", "贴身制服", "外套")
		IF GETBIT(TEQUIP:野外PLAY, 0)
			SOURCE:0:露出 += 4000
			SOURCE:0:逸脱 += 500
		ELSE
			SOURCE:0:露出 += 1000
			SOURCE:0:逸脱 += 200
		ENDIF
	ENDIF
ELSEIF CHECK_CLO("没穿内褲")
	;種族衣装に内衣有
	IF CLO_ORI(TARGET, "胖次")
		IF GETBIT(TEQUIP:野外PLAY, 0)
			SOURCE:0:露出 += 2000
			SOURCE:0:逸脱 += 300
		ELSE
			SOURCE:0:露出 += 600
			SOURCE:0:逸脱 += 100
		ENDIF
	ENDIF
ENDIF

;昼だと恥ずかしい
IF COND("夜") == 0
	TIMES SOURCE:0:露出, 2.00
	TIMES SOURCE:0:逸脱, 2.00
ENDIF

;PALAM:欲情をみる
SELECTCASE PALAM:欲情
CASE IS < PALAMLV:1
	TIMES SOURCE:0:露出, 0.80
CASE IS < PALAMLV:2
	TIMES SOURCE:0:露出, 0.90
CASE IS < PALAMLV:3
	TIMES SOURCE:0:露出, 1.00
CASE IS < PALAMLV:4
	TIMES SOURCE:0:露出, 1.10
CASE IS >= PALAMLV:4
	TIMES SOURCE:0:露出, 1.20
ENDSELECT
;ABL:受虐属性をみる
SELECTCASE ABL:受虐属性
CASE 0
	TIMES SOURCE:0:露出, 0.80
CASE 1
	TIMES SOURCE:0:露出, 1.00
CASE 2
	TIMES SOURCE:0:露出, 1.30
CASE 3
	TIMES SOURCE:0:露出, 1.60
CASE 4
	TIMES SOURCE:0:露出, 2.00
CASEELSE
	TIMES SOURCE:0:露出, 3.00
ENDSELECT

SIF TALENT:喜歡引人注目
	TIMES SOURCE:0:露出, 1.50
SIF TALENT:内気
	TIMES SOURCE:0:逸脱, 2.00
SIF TALENT:害羞
	TIMES SOURCE:0:逸脱, 2.00

SOURCE:0:恐惧 += SOURCE:0:逸脱 / 2

CALL CALC, "調教ソース", ARG

@EQUIP_COM171
SIF TEQUIP:野外PLAY == 0
	RETURN 0

CALL SOURCE_COM171, 100

@EQUIP_COM171_2
SIF TEQUIP:野外PLAY == 0
	RETURN 0
;露出経験の値を管理
LOCAL = 1
IF GETBIT(TEQUIP:野外PLAY, 0)
	PRINTL ＜野外PLAY中(人里)＞
	IF COND("夜") == 0
		LOCAL = 3
	ELSE
		LOCAL = 2
	ENDIF
ELSEIF GETBIT(TEQUIP:野外PLAY, 1)
	PRINTL ＜野外PLAY中(森林)＞

	;獣耳ならBOUNS
	SIF PLAYER == MASTER && TALENT:動物耳
		TFLAG:好感度BOUNS += 1
	;一部種族にBOUNS
	SIF BASE:種族 == 3
		TFLAG:好感度BOUNS += 1
ELSEIF GETBIT(TEQUIP:野外PLAY, 2)
	PRINTL ＜野外PLAY中(花園)＞

	;花園補正。好感度にBOUNS。
	SIF PLAYER == MASTER
		TFLAG:好感度BOUNS += 1
ELSEIF GETBIT(TEQUIP:野外PLAY, 3) || GETBIT(TEQUIP:野外PLAY, 4)
	PRINTL ＜野外PLAY中(水辺)＞

	;夏ならBOUNS
	SIF PLAYER == MASTER && FLAG:季節 == 1
		TFLAG:好感度BOUNS += 2

	;水辺補正。汚れを洗い流せる。
	;汚れを初期化
	CALL RESET_STAIN_EX, "全部", MASTER
	CALL RESET_STAIN_EX, "全部", TARGET
	;助手がいる場合は、助手もきれいに
	SIF ASSI
		CALL RESET_STAIN_EX, "全部", ASSI
ENDIF

EXP:露出経験 += LOCAL

SIF PLAYER == MASTER && ABL:露出癖 >= 3
	TFLAG:好感度BOUNS += 1


code:
@COM172
;浴室PLAY
CALL PRINT_TRAIN_NAME(SELECTCOM)

CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

;浴室場への出入り
CALL SETFLAG, "浴室PLAY"

RETURN 1

@FLAG_COM172, ARG

@SOURCE_COM172, ARG
CALL VARSET_COMF

SOURCE:0:露出 = 200

;PALAM:欲情をみる
SELECTCASE PALAM:欲情
CASE IS < PALAMLV:1
	TIMES SOURCE:0:露出, 0.80
CASE IS < PALAMLV:2
	TIMES SOURCE:0:露出, 0.90
CASE IS < PALAMLV:3
	TIMES SOURCE:0:露出, 1.00
CASE IS < PALAMLV:4
	TIMES SOURCE:0:露出, 1.10
CASE IS >= PALAMLV:4
	TIMES SOURCE:0:露出, 1.20
ENDSELECT

CALL CALC, "調教ソース", ARG

@TRAIN_MESSAGE_COM172
IF TEQUIP:浴室PLAY
	PRINTFORMW 和%CALLNAME:TARGET%回到部屋了……
ELSE
	PRINTFORM %CALLNAME:TARGET%
	IF CHECK_CLO("泳装姿")
		PRINTFORM 就那樣穿著泳衣
		CALL SET_SHIRT, ""
		CALL SET_OUTER, ""
		CALL SET_SOCKS, ""
		CALL SET_GLOVE, ""
	ELSEIF CHECK_CLO("全裸") == 0
		PRINTFORM 在更衣室脫掉衣服後
		CALL SETFLAG, "衣装全脱ぎ", TARGET
	ENDIF
	PRINTFORMW 一起向浴室走去……
	IF TEQUIP:眼罩
		PRINTFORM 被眼罩遮住雙眼的%CALLNAME:TARGET%、將%CALLNAME:PLAYER%
		IF TALENT:坦率 || TALENT:内気
			PRINTFORM 的手牽起、戰戰兢兢地
		ELSE
			PRINTFORM 的手腕摸索著抓住、為了防止滑倒而小心地
		ENDIF
		PRINTFORMW 走進了浴室……
	ELSEIF TALENT:害羞 && ABL:露出癖 < 3 && CHECK_CLO("裸体")
		PRINTFORMW 進入浴室的%CALLNAME:TARGET%臉色羞紅、用雙手徒勞的遮掩著美麗的身體……
	ELSEIF TALENT:献身 && (TALENT:好色 || TALENT:淫乱)
		PRINTFORMW 進入浴室的%CALLNAME:TARGET%浮現追逐快樂的笑容、拿起肥皂潤滑著身體……
	ELSEIF TALENT:献身
		PRINTFORMW 進入浴室的%CALLNAME:TARGET%為了幫助%CALLNAME:MASTER%擦洗而拿起了角落的浴桶……
	ELSEIF (TALENT:好色 || TALENT:淫乱 || TALENT:小悪魔 || TALENT:坏心眼) && TALENT:害羞 == 0 && CHECK_CLO("裸体")
		PRINTFORMW 進入浴室的%CALLNAME:TARGET%對美麗的身體完全不做遮掩、反而擺出誘惑的姿勢、向%CALLNAME:MASTER%露出%TEXTR("挑逗的/大膽的")%笑容……
	ELSEIF TALENT:不知羞恥 && CHECK_CLO("裸体")
		PRINTFORMW 進入浴室的%CALLNAME:TARGET%即使美麗的嬌軀完全的暴露在%CALLNAME:MASTER%眼前、也完全沒有一絲羞澀的表現……
	ENDIF
ENDIF

@EQUIP_COM172
SIF TEQUIP:浴室PLAY == 0
	RETURN 0
SIF SELECTCOM == 172
	RETURN 0
;浴室PLAY
CALL SOURCE_COM172, 30

RETURN 1

@EQUIP_COM172_2
SIF TEQUIP:浴室PLAY == 0
	RETURN 0
SIF SELECTCOM == 172
	RETURN 0

PRINTL ＜浴室PLAY中＞

TFLAG:好感度BOUNS += 1

RETURN 1


code:
@COM173
;新妻PLAY
CALL PRINT_TRAIN_NAME(SELECTCOM)

CALL TRAIN_MESSAGE_B

SIF REFUSE_CHECK()
	RETURN 0

STR:前回指令 = %STR:今回指令%

;圍裙のセット
;終了時は修正无
IF TEQUIP:新妻PLAY
	TEQUIP:新妻PLAY = 0
	RETURN 1
ELSE
	TEQUIP:新妻PLAY = 1
ENDIF

RETURN 1

@FLAG_COM173, ARG

@SOURCE_COM173, ARG
CALL VARSET_COMF

SOURCE:0:情愛 = 500
SOURCE:0:逸脱 = 100

;ABL:奉仕精神をみる
SELECTCASE ABL:奉仕精神
CASE 0
CASE 1
	TIMES SOURCE:0:情愛, 1.20
CASE 2
	TIMES SOURCE:0:情愛, 1.40
CASE 3
	TIMES SOURCE:0:情愛, 1.60
CASE 4
	TIMES SOURCE:0:情愛, 1.80
CASEELSE
	TIMES SOURCE:0:情愛, 2.00
ENDSELECT

SIF TALENT:好奇心
	TIMES SOURCE:0:情愛, 1.50
SIF TALENT:献身
	TIMES SOURCE:0:情愛, 1.50
SIF TALENT:内気
	TIMES SOURCE:0:逸脱, 1.50
SIF TALENT:害羞
	TIMES SOURCE:0:逸脱, 1.20
SIF TALENT:高慢
	TIMES SOURCE:0:逸脱, 2.00

CALL CALC, "調教ソース", ARG

@TRAIN_MESSAGE_COM173
IF TEQUIP:新妻PLAY
	PRINTFORMW 讓%CALLNAME:TARGET%將圍裙脫掉了……
ELSE
	PRINTFORM %CALLNAME:TARGET%
	IF CHECK_CLO("裸体")
		PRINTFORMW 穿上了裸體圍裙……
	ELSE
		PRINTFORMW 穿上了圍裙……
	ENDIF
	;相思相愛
	IF TALENT:相思相愛 && PLAYER == MASTER && !CHECK_CLO("裸体")
		PRINTFORMW %CALLNAME:TARGET%似乎非常高興、向%CALLNAME:MASTER%露出了美麗的笑容……
	ELSEIF ((TALENT:恋慕 && PLAYER == MASTER) || TALENT:容易迷恋) && !CHECK_CLO("裸体")
		PRINTFORMW %CALLNAME:TARGET%的臉頰因羞澀而染上紅暈、扭捏地看向%CALLNAME:MASTER%……
	ELSEIF (TALENT:不知羞恥 || TALENT:好色) && CHECK_CLO("裸体")
		PRINTFORMW %CALLNAME:TARGET%對此似乎漠不關心、只是用渴求快樂的眼神看過來……
	ELSEIF (TALENT:小悪魔 || TALENT:傲慢) && !CHECK_CLO("裸体")
		PRINTFORMW %CALLNAME:TARGET%好奇的打量著現在的自己、眼神閃亮的捲起了下擺……
	ELSEIF (TALENT:坦率 || COND("種族：人形")) && !CHECK_CLO("裸体")
		PRINTFORMW %CALLNAME:TARGET%在換上圍裙後就向廚房走去、%CALLNAME:MASTER%阻止了她……
	ENDIF
ENDIF

@EQUIP_COM173
SIF TEQUIP:新妻PLAY == 0
	RETURN 0
SIF SELECTCOM == 173
	RETURN 0
;新妻PLAY
CALL SOURCE_COM173, 30

RETURN 1

@EQUIP_COM173_2
SIF TEQUIP:新妻PLAY == 0
	RETURN 0
SIF SELECTCOM == 173
	RETURN 0
;新妻PLAY
PRINTL ＜新妻PLAY中＞

SIF ASSIPLAY == 0 && ABL:信頼 >= 3
	TFLAG:好感度BOUNS += 2
SIF ASSIPLAY == 0 && TALENT:恋慕
	TFLAG:好感度BOUNS += 5

RETURN 1